


IL PRIMO SETTIMANALE DI SOFTWARE SU CARTA

L. 1.000

PER IL TUO PERSONAL COMPUTER

Una pubblicazione della J.soft editrice



# PAPER soft



Anno 2 · N° 6 · 15 febbraio 1985



**Labirinto tridimensionale**  
**Calendario perpetuo**



**Tic Tac Toe**  
**Attack!**



**Capogiro**  
**Grandi stampe**



**Caccia al tesoro**



**Puleino**  
**Guitar**

Editrice

**J.soft**

Con la collaborazione del Gruppo Editoriale Jackson

# Guida all'input dei programmi sullo ZX Spectrum

Le "parole" comprese tra parentesi graffe indicano i caratteri grafici predefiniti (G), il tasto (numero, seguente la G), la necessità di premere il tasto insieme a CAPS SHIFT (eventuale S precedente la G) e il numero di ripetizioni del tasto richieste (eventuale numero all'inizio della "parola"). I caratteri grafici definiti dal programma sono invece indicati da lettere maiuscole (corrispondenti al tasto da usare) sottolineate.

Quando leggete	Premete	Vedrete
{G1}	1	.
{G2}	2	..
{G3}	3	...
{G4}	4	....
{G5}	5	.....
{G6}	6	.....
{G7}	7	.....
{G8}	8	.....
{SG1}	CAPS SHIFT 1	..
{SG2}	CAPS SHIFT 2	..
{SG3}	CAPS SHIFT 3	..
{SG4}	CAPS SHIFT 4	..
{SG5}	CAPS SHIFT 5	..
{SG6}	CAPS SHIFT 6	..
{SG7}	CAPS SHIFT 7	..
{SG8}	CAPS SHIFT 8	..

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

Quando leggete	Premete	Vedrete
A	A	Simbolo grafico definito nel programma in uso.
B	B	
C	C	
D	D	
E	E	
F	F	
G	G	
H	H	
I	I	
J	J	
K	K	
L	L	
M	M	
N	N	
O	O	
P	P	
Q	Q	
R	R	
S	S	
T	T	
U	U	


# Guida per l'input dei programmi versioni VIC 20 e C64

Notate che i listati contengono "parole" racchiuse tra parentesi graffe { }. Tali parole rappresentano caratteri di controllo come mostrato nel sottostante riquadro. Se sono precedute da un numero, questo indica il numero di volte che quel tasto deve essere premuto. Se il simbolo è sottolineato deve essere premuto contemporaneamente a SHIFT mentre se è racchiuso da [ < > ] deve essere

premutato contemporaneamente al tasto COMMODORE. Inoltre, se tra parentesi si trova un carattere alfabetico "solitario", questo dovrà essere premuto contemporaneamente al tasto CONTROL. Con questo sistema di codifica, sarà molto più agevole copiare i listati senza faticose e dubbie interpretazioni di caratteri grafici e di controllo del cursore o dei colori.

{CLR}	SHIFT CLR/HOME	⬅	{CYN}	CTRL 4	⬇	[<7>]	G 7	⬆
{HOME}	CLR/HOME	⬅	{PUR}	CTRL 5	⬇	[<8>]	G 6	⬆
{SU} ~	SHIFT    CRSR	⬅	{GRN}	CTRL 6	⬆	{F1}	F1	⬆
{GIU'}	CRSR	⬅	{BLU}	CTRL 7	⬆	{F2}	F2	⬆
{SIN}	SHIFT ⇐ CRSR ⇒	⬅	{YEL}	CTRL 6	⬆	{F3}	F3	⬆
{DES}	⇐ CRSR ⇒	⬅	[<1>]	G 1	⬆	{F4}	F4	⬆
{RVS}	CTRL 9	⬅	[<2>]	G 2	⬆	{F5}	F5	⬆
{OFF}	CTRL 8	⬅	[<3>]	G 3	⬆	{F6}	F6	⬆
{BLK}	CTRL 1	⬅	[<4>]	G 4	⬆	{F7}	F7	⬆
{WHT}	CTRL 2	⬅	[<5>]	G 5	⬆	{F8}	F8	⬆
{RED}	CTRL 3	⬅	[<6>]	G 6	⬆			⬆





# PAPER SOFT

	<b>4</b>	<b>Labirinto tridimensionale</b> di D. Ward trad. e adatt. di M. Cerofolini
	<b>8</b>	<b>Calendario perpetuo</b> di M. Effron trad. e adatt. di M. Cerofolini
Extended basic 	<b>11</b>	<b>Tic Tac Toe</b> di M. Bianchi
Extended basic 	<b>14</b>	<b>Attack!</b> di H. Renko trad. e adatt. di F. Re Garbagnati
16/48K 	<b>15</b>	<b>Capogiro</b> di G.P. Grandin trad. e adatt. di C. Panzalis
16/48K 	<b>18</b>	<b>Grandi stampe</b> di J. Southgate trad. e adatt. di C. Panzalis
Joystick 	<b>19</b>	<b>Caccia al tesoro</b> di C. Overhulser trad. e adatt. di F. Sarcina
	<b>23</b>	<b>Pulcino</b> di S. Swap trad. e adatt. di F. Sarcina
	<b>25</b>	<b>Guitar</b> di M. Ravagnan

J. soft s.r.l.

#### DIREZIONE, REDAZIONE, AMMINISTRAZIONE

V.le Restelli, 5  
20124 MILANO  
Tel. (02) 68.88.228-68.37.97

#### DIRETTORE RESPONSABILE:

Pietro Dell'Orco

#### COORDINAMENTO TECNICO:

Riccardo Paolillo

#### REDAZIONE:

Lucio Bragagnolo  
Mauro Cristub. Grizzi

#### GRAFICA E IMPAGINAZIONE:

Margherita La Noce  
Ivana Rossi  
Raffaella Toffolatti

#### FOTOCOMPOSIZIONE:

d&b Via Vignola, 5  
Tel. 02/59.85.08  
20133 MILANO

#### CONTABILITÀ:

Giulia Pedrazzini  
Flavia Bonatti

#### AUTORIZZAZIONE ALLA PUBBLICAZIONE:

Tribunale di Milano n° 200  
del 14.04.1984

#### STAMPA:

Elcograf - Beverate (CO)



Rivista associata  
all'Unione stampa  
Periodica Italiana

#### PUBBLICITÀ

Concessionario per l'Italia e l'Estero  
J.Advertising s.r.l.  
V.le Restelli, 5  
20124 MILANO  
Tel. (02) 68.82.895-68.80.606-68.87.233  
Tlx. 316213 REINA I  
Concessionario esclusivo per la  
DIFFUSIONE in Italia e Estero:  
SODIP - Via Zuretti, 25  
20125 MILANO

\* Spedizione in abbonamento  
postale Gruppo II/70  
Prezzo della rivista L. 1.000  
Numeri arretrati L. 2.000

© TUTTI I DIRITTI DI  
RIPRODUZIONE O TRADUZIONE  
DEGLI ARTICOLI PUBBLICATI  
SONO RISERVATI

# Labirinto tridimensionale



Eccovi un labirinto con un insolito punto di vista: quello di un verme!! Invece della solita vista dall'alto questo gioco vi permetterà di cercare la via di uscita per tentativi con la prospettiva di una persona effettivamente all'interno del labirinto. Si può scegliere di lasciare delle impronte che mostrano la direzione verso la quale si sta andando o chiedere di avere una vista dall'alto del labirinto per vedere ef-

fettivamente il punto in cui ci si trova. Inoltre il programma prevede dieci diversi livelli di difficoltà impegnando sia il pivellino che il perfetto piccione viaggiatore.

Un consiglio prima di iniziare: le prime volte usate molto la mappa per capire come viene rappresentata la prospettiva del labirinto e controllate sempre la bussola che vi indicherà in quale direzione siete orientati.

```

10  FOR X = 770 TO 790
20  READ Y
30  POKE X,Y
40  NEXT X
90  DEF FN R(X) = INT ( RND (1)
    * X)
100  DIM C(3121):F = 1
110  GOTO 1440
120  REM
130  GR :MC = 3
140  COLOR= 3:A = FN R(19):B = FN
    R(19):X = A * 2 + 1:Y = 8 *
    2 + 1: PLOT X,Y:SC = 360
150  IF FN R(100) < S * 3 THEN 2
    60
160  L = 0: IF A < 1 THEN 170
165  IF SCRN( X - 2,Y) = 0 THEN
    L = 1
170  R = 0: IF A > 17 THEN 180
175  IF SCRN( X + 2,Y) = 0 THEN
    R = 1
180  U = 0: IF B < 1 THEN 190
185  IF SCRN( X,Y - 2) = 0 THEN
    , U = 1
190  D = 0: IF B > 17 THEN 200
195  IF SCRN( X,Y + 2) = 0 THEN
    D = 1
200  IF L + R + U + D = 0 THEN 26
    0
210  IF FN R(4) = 0 GOTO 220
211  IF FN R(4) = 1 GOTO 230
212  IF FN R(4) = 2 GOTO 240
213  IF FN R(4) = 3 GOTO 250
220  IF L = 0 THEN 210
225  A = A - 1:X = X - 2:SC = SC -
    1: HLIN X,X + 1 AT Y: GOTO 1
    50
230  IF R = 0 THEN 210
235  A = A + 1:X = X + 2:SC = SC -
    1: HLIN X - 1,X AT Y: GOTO 1
    50
240  IF U = 0 THEN 210
245  B = B - 1:Y = Y - 2:SC = SC -
    1: VLIN Y,Y + 1 AT X: GOTO 1
    50

```

```

250  IF D = 0 THEN 210
255  B = B + 1:Y = Y + 2:SC = SC -
    1: VLIN Y - 1,Y AT X: GOTO 1
    50
260  IF SC = 0 THEN 1550
270  A = FN R(19):B = FN R(19):X
    = A * 2 + 1:Y = B * 2 + 1: IF
    SCRN( X,Y) = 0 THEN 270
275  GOTO 160
280  RETURN
290  REM
300  COLOR= 7: FOR I = 1 TO 18
310  HLIN 19 - I,19 + 1 AT 19 - I

320  NEXT I
330  REM
340  COLOR= 6
350  FOR I = 1 TO 19
360  HLIN 20 - I,18 + 1 AT 18 + I

370  NEXT I
380  REM
390  COLOR= 12
400  FOR I = 1 TO 18
410  REM
420  VLIN 19 - I,18 + 1 AT 19 - I

430  REM
440  VLIN 19 - I,18 + 1 AT 19 + I

450  NEXT I
460  GOSUB 1230: IF MAP < > 1 THEN
    470
465  CALL - 936: GOSUB 1230: PRINT
    "PREMERE 'M' PER RICHIEDERE
    LA MAPPA"
470  RETURN
480  REM

490  IF I = 0 GOTO 500
491  IF I = 1 GOTO 510
492  IF I = 2 GOTO 520
493  IF I = 3 GOTO 530
494  IF I = 4 GOTO 540
495  IF I = 5 GOTO 550

```

```

500 FOR Q = 1 TO 5: COLOR= 4: VLIN
    Q,38 - Q AT 38 - Q: COLOR= 7
    : HLIN 37 - Q,37 AT Q: COLOR=
    6: HLIN 37 - Q,37 AT 38 - Q:
    NEXT Q: RETURN
510 FOR Q = 6 TO 10: COLOR= 4: VLIN
    Q,38 - Q AT 38 - Q: COLOR= 7
    : HLIN 28,32 AT Q: COLOR= 6:
    HLIN 28,32 AT 38 - Q: NEXT
    Q: RETURN
520 FOR Q = 11 TO 14: COLOR= 4: VLIN
    Q,38 - Q AT 38 - Q: COLOR= 7
    : HLIN 24,27 AT Q: COLOR= 6:
    HLIN 24,27 AT 38 - Q: NEXT
    Q: RETURN
530 FOR IN = 21 TO 22: COLOR= 4:
    VLIN 17,20 AT IN: NEXT IN
535 COLOR= 7: PLOT 22,16: COLOR=
    6: PLOT 22,21: RETURN
540 COLOR= 4: PLOT 20,19: COLOR=
    7: PLOT 20,18: COLOR= 6: PLOT
    20,20: RETURN
550 COLOR= 4: PLOT 19,19: RETURN

560 REM
570 IF I = 0 GOTO 580
571 IF I = 1 GOTO 590
572 IF I = 2 GOTO 600
573 IF I = 3 GOTO 610
574 IF I = 4 GOTO 620
575 IF I = 5 GOTO 630
580 FOR Q = 1 TO 5: COLOR= 4: VLIN
    Q,38 - Q AT Q: COLOR= 7: HLIN
    1,Q AT Q: COLOR= 6: HLIN 1,Q
    AT 38 - Q: NEXT Q: RETURN
590 FOR Q = 6 TO 10: COLOR= 4: VLIN
    Q,38 - Q AT Q: COLOR= 7: HLIN
    6,Q AT Q: COLOR= 6: HLIN 6,Q
    AT 38 - Q: NEXT Q: RETURN
600 FOR Q = 11 TO 14: COLOR= 4: VLIN
    Q,38 - Q AT Q: COLOR= 7: HLIN
    11,Q AT Q: COLOR= 6: HLIN 11
    ,Q AT 38 - Q: NEXT Q: RETURN

610 FOR IN = 16 TO 17: COLOR= 4:
    VLIN 17,20 AT IN: NEXT IN
615 COLOR= 7: PLOT 16,16: COLOR=
    6: PLOT 16,21: RETURN
620 COLOR= 4: PLOT 18,19: COLOR=
    7: PLOT 18,18: COLOR= 6: PLOT
    18,20: RETURN
630 COLOR= 4: PLOT 19,19: RETURN

640 REM
650 COLOR= 4: IF Z < > 201 THEN
    660
655 FOR Q = 1 TO 37: HLIN 1,37 AT
    Q: NEXT Q: RETURN
660 IF Z < > 211 THEN 670
665 FOR Q = 1 TO 37: VLIN 1,37 AT
    Q: NEXT Q: RETURN
670 IF Z < > 196 THEN RETURN
675 FOR Q = 1 TO 37: VLIN 1,37 AT
    38 - Q: NEXT Q: RETURN
680 REM
690 COLOR= 4: IF C(K) = 15 THEN
    COLOR= 15
691 IF I = 0 GOTO 700

692 IF I = 1 GOTO 710
693 IF I = 2 GOTO 720
694 IF I = 3 GOTO 730
695 IF I = 4 GOTO 740
696 IF I = 5 GOTO 750
700 FOR IN = 2 TO 35: VLIN 2,35 AT
    IN: NEXT IN: I = 6: RETURN
710 FOR IN = 6 TO 32: VLIN 6,32 AT
    IN: NEXT IN: I = 6: RETURN
720 FOR IN = 11 TO 27: VLIN 11,2
    7 AT IN: NEXT IN: I = 6: RETURN
730 FOR IN = 15 TO 23: VLIN 15,2
    3 AT IN: NEXT IN: I = 6: RETURN
740 FOR IN = 18 TO 20: VLIN 17,2
    0 AT IN: NEXT IN: I = 6: RETURN

750 PLOT 19,19: I = 6: RETURN
760 REM
770 FOR I = 0 TO 5
780 TX = X - 1: TY = Y + I: IF TX <
    1 OR TY > 37 THEN 800
785 K = 39 * TX + 41 * TY
790 IF C(K) = MC OR C(K) = TC THEN
    GOSUB 490
800 TX = X + 1: IF TX < 1 THEN 82
    0
805 K = 39 * TX + 41 * TY
810 IF C(K) = MC OR C(K) = TC THEN
    GOSUB 570
820 TX = X: TY = Y + I: IF TY > 38
    THEN 850
825 K = 39 * TX + 41 * TY
830 IF C(K) < > MC AND C(K) < >
    TC THEN GOSUB 690
840 IF C(K) = TC AND FTS = 1 THEN
    GOSUB 1270
850 NEXT I: RETURN
860 REM
870 FOR I = 0 TO 5
880 TY = Y - 1: TX = X - I: IF TX <
    1 OR TY < 1 THEN 900
885 K = 39 * TX + 41 * TY
890 IF C(K) = MC OR C(K) = TC THEN
    GOSUB 490
900 TY = Y + 1: IF TY > 37 THEN 9
    20
905 K = 39 * TX + 41 * TY
910 IF C(K) = MC OR C(K) = TC THEN
    GOSUB 570
920 TY = Y: TX = X - 1: IF TX < 0 THEN
    950
925 K = 39 * TX + 41 * TY
930 IF C(K) < > MC AND C(K) < >
    TC THEN GOSUB 690
940 IF C(K) = TC AND FTS = 1 THEN
    GOSUB 1270
950 NEXT I: RETURN
960 REM
970 FOR I = 0 TO 5
980 TY = Y + 1: TX = X + I: IF TX >
    37 OR TY > 37 THEN 1000
985 K = 39 * TX + 41 * TY
990 IF C(K) = MC OR C(K) = TC THEN
    GOSUB 490
1000 TY = Y - 1: IF TY < 1 THEN 1
    020

```

```

1005 K = 39 * TX + 41 * TY
1010 IF C(K) = MC OR C(K) = TC THEN
    GOSUB 570
1020 TY = Y:TX = X + 1: IF TX > 3
    8 THEN 1050
1025 K = 39 * TX + 41 * TY
1030 IF C(K) < ) MC AND C(K) <
    > TC THEN GOSUB 690
1040 IF C(K) = TC AND FTS = 1 THEN
    GOSUB 1270
1050 NEXT 1: RETURN
1060 REM
1070 FOR 1 = 0 TO 5
1080 TX = X + 1:TY = Y - 1: IF TX
    > 37 OR TY < 1 THEN 1100
1085 K = 39 * TX + 41 * TY
1090 IF C(K) = MC OR C(K) = TC THEN
    GOSUB 490
1100 TX = X - 1: IF TX < 1 THEN 1
    120
1105 K = 39 * TX + 41 * TY
1110 IF C(K) = MC OR C(K) = TC THEN
    GOSUB 570
1120 TX = X:TY = Y - 1: IF TY < 0
    THEN 1150
1125 K = 39 * TX + 41 * TY
1130 IF C(K) < ) MC AND C(K) <
    > TC THEN GOSUB 690
1140 IF C(K) = TC AND FTS = 1 THEN
    GOSUB 1270
1150 NEXT 1: RETURN
1160 REM
1170 GR
1180 XX = X:YY = Y:SC = SC + 5
1190 FOR X = 0 TO 38: FOR Y = 0 TO
    38
1200 K = 39 * X + 41 * Y: COLOR=
    C(K): PLOT X,Y
1210 NEXT Y,X:X = XX:Y = YY: FOR
    1 = 1 TO 1000: NEXT 1: GOSUB
    290: GOTO 1810
1220 REM
1230 IF COMP < ) 1 THEN RETURN
1240 H$ = "NORD EST SUD OVEST":
    CALL - 936: VTAB 24
1250 HTAB 17: PRINT MID$ (H$,(D
    1R + 1) * 5 - 4,5): RETURN
1260 REM
1270 IF FTS < ) 1 THEN RETURN
1280 IF C(K) < ) TC THEN RETURN
1285 F = - F
1290 COLOR= 0: PLOT 19 + F,37 -
    1 * 3
1300 RETURN
1310 REM
1320 FOR A = 0 TO 38
1330 FOR B = 0 TO 38
1340 K = 39 * A + 41 * B: IF A =
    19 AND B = 19 THEN GOSUB 14
    10
1350 C(K) = SCRN( A,B): NEXT B,A
    : RETURN
1360 REM
1370 REM
1390 RETURN

```

```

1400 REM
1410 PRINT SPC( 4);" 'A'=AVANTI
    'S'=SINISTRA
    'D'=DESTRA '1'=IND
    IETRO"
1420 FOR I = 1 TO 2000: NEXT 1: RETURN
1430 REM
1440 TEXT : CALL - 936: VTAB 9
1450 HTAB 13: VTAB 3: POKE 50,63
    : PRINT " LABIRINTO ": POKE
    50,255
1460 FOR D = 1 TO 1000: NEXT D: POKE
    - 1629B,0
1480 HTAB 1: VTAB 7: PRINT "SI P
    OSSONO RICHIEDERE I SEGUENTI
    AIUT1:"
1500 HTAB 2: VTAB 10: INPUT "IMP
    RONTE (S/N) ? ";G$:FTS = 0: IF
    MID$ (G$,1,1) = "S" THEN FT
    S = 1
1510 HTAB 2: VTAB 13: INPUT "BUS
    SOLA (S/N) ? ";G$:COMP = 0:
    IF MID$ (G$,1,1) = "S" THEN
    COMP = 1
1520 HTAB 2: VTAB 16: INPUT "MAP
    PA (SI PERDONO 5 PUNTI OGNI
    VOLTA CHE SI USA)
    (S/N) ? ";G$:MAP = 0: IF MID$
    (G$,1,1) = "S" THEN MAP = 1
1540 VTAB 21: HTAB 9: INPUT "DIF
    FICOLTA' ?(1-10) ";S$: IF S$
    < "1" OR S$ > "10" THEN 154
    0
1542 CALL - 936: VTAB 24: HTAB
    7: PRINT "CREAZIONE LABIRINT
    O IN CORSO": GOSUB 130
1550 CALL - 936
1560 REM
1570 PRINT " PARTENZA=ROSSO
    ARRIVO=BIANCO":TC = 1:L =
    FN R(4): IF L < ) 0 THEN 1
    580
1575 COLOR= 15: PLOT 1,37: COLOR=
    TC:X = 37:Y = 1: PLOT X,Y:E =
    1:G = 37: GOTO 1610
1580 IF L < ) 1 THEN 1590
1585 COLOR= 15: PLOT 1,1:X = 37:
    Y = 37: COLOR= TC: PLOT X,Y:
    E = 1:G = 1: GOTO 1610
1590 IF L < ) 2 THEN 1600
1595 COLOR= 15: PLOT 37,1:X = 1:
    Y = 37: COLOR= TC: PLOT X,Y:
    E = 37:G = 1: GOTO 1610
1600 COLOR= 15: PLOT 37,37:X = 1
    :Y = 1: COLOR= TC: PLOT X,Y:
    E = 37:G = 37: GOTO 1610
1610 GOSUB 1320: GOSUB 1370
1620 POKE - 1636B,0
1630 DIR = FN R(4): GOSUB 290: GOTO
    1810
1640 REM
1650 Z = PEEK ( - 16384): IF Z <
    127 THEN 1650
1655 XX = 3 * ( ABS (X - E) + ABS
    (Y - G)): POKE 76B,XX: POKE
    769,25: CALL 770
1660 REM
1670 IF Z = 193 AND DIR > = 4 THEN

```



```

DIR = DIR - (4 * INT (DIR /
4))
1672 IF Z = 196 AND (DIR + 1) >
= 4 THEN DIR = (DIR + 1) -
(4 * INT ((DIR + 1) / 4)): GOTO
1680
1673 IF Z = 196 AND (DIR + 1) <
4 THEN DIR = DIR + 1
1674 IF Z = 201 AND (DIR + 2) >
= 4 THEN DIR = (DIR + 2) -
(4 * INT ((DIR + 2) / 4)): GOTO
1680
1675 IF Z = 201 AND (DIR + 2) <
4 THEN DIR = DIR + 2
1676 IF Z = 211 AND (DIR + 3) >
= 4 THEN DIR = (DIR + 3) -
(4 * INT ((DIR + 3) / 4)): GOTO
1680
1677 IF Z = 211 AND (DIR + 3) <
4 THEN DIR = DIR + 3
1680 IF MAP ( ) 1 THEN 1690
1685 IF Z = 205 THEN GOTO 1170
1690 IF Z = 193 THEN 1700
1695 GOSUB 640: GOSUB 290: GOTO
1810
1700 GOSUB 290: GOSUB 1360
1710 REM
1720 IF DIR ( ) > 0 THEN 1740
1730 Y = Y - 1: K = 39 * X + 41 *
Y: IF C(K) = 0 OR C(K) = 1 THEN
SC = SC + 1
1735 IF C(K) = 0 THEN Y = Y + 1
1740 IF DIR ( ) > 2 THEN 1760
1750 Y = Y + 1: K = 39 * X + 41 *
Y: IF C(K) = 0 OR C(K) = 1 THEN
SC = SC + 1
1755 IF C(K) = 0 THEN Y = Y - 1
1760 IF DIR ( ) > 3 THEN 1780
1770 X = X - 1: K = 39 * X + 41 *
Y: IF C(K) = 0 OR C(K) = 1 THEN
SC = SC + 1
1775 IF C(K) = 0 THEN X = X + 1
1780 IF DIR ( ) 1 THEN 1800
1790 X = X + 1: K = 39 * X + 41 *
Y: IF C(K) = 0 OR C(K) = 1 THEN
SC = SC + 1
1795 IF C(K) = 0 THEN X = X - 1
1800 K = 39 * X + 41 * Y: IF C(K)
= 15 THEN 1850
1810 IF DIR = 0 THEN QWQ = 1070
1812 IF DIR = 1 THEN QWQ = 970
1814 IF DIR = 2 THEN QWQ = 770
1816 IF DIR = 3 THEN QWQ = 870
1820 IF Z = 193 THEN GOTO 1830
1821 POKE - 16368,0: IF QWQ = 1
070 THEN GOSUB 1070
1822 IF QWQ = 970 THEN GOSUB 97
0
1823 IF QWQ = 770 THEN GOSUB 77
0
1824 IF QWQ = 870 THEN GOSUB 87
0
1825 GOTO 1650
1830 C(K) = TC
1831 IF QWQ = 1070 THEN GOSUB 1
070
1832 IF QWQ = 970 THEN GOSUB 97
0

```

```

1833 IF QWQ = 770 THEN GOSUB 77
0
1834 IF QWQ = 870 THEN GOSUB 87
0
1839 POKE - 16368,0: GOTO 1650
1840 REM
1850 COLOR= 15: FOR Q = 0 TO 17:
HLIN 2,36 AT 19 + Q: HLIN 2
,36 AT 19 - Q: NEXT Q: POKE
- 16368,0: CALL - 936
1860 COLOR= 9: HLIN 16,19 AT 18:
HLIN 16,18 AT 20: HLIN 16,1
9 AT 22: HLIN 22,23 AT 18: HLIN
21,24 AT 20: VLIN 20,22 AT 1
2: VLIN 18,22 AT 16: VLIN 19
,22 AT 21
1870 VLIN 19,22 AT 24: VLIN 17,2
0 AT 27: PLOT 10,18: PLOT 11
,19: PLOT 14,18: PLOT 13,19:
PLOT 27,22: GOSUB 1360: GOSUB
1910
1880 FOR D = 1 TO 4000: NEXT D
1890 GOTO 1440
1900 REM
1910 POKE 768,250: POKE 769,50: CALL
770
1920 POKE 768,200: POKE 769,50: CALL
770
1930 POKE 768,160: POKE 769,50: CALL
770
1940 POKE 768,130: POKE 769,120:
CALL 770
1950 POKE 768,160: POKE 769,50: CALL
770
1960 POKE 768,130: POKE 769,255:
CALL 770
1970 REM
1980 HTAB 10: PRINT "PUNTEGGIO =
";(100 - SC)
1990 RETURN
2000 DATA 173,48,192,136,208,5,2
06,1,3,240,9,202,208,245,174
,0,3,76,2,3,96

```

## Guida per l'input dei programmi sul TI 99/4A

A causa del sistema di codifica utilizzato, possono capitare in alcune linee delle improvvise "andate a capo" delle quali, naturalmente, non va tenuto conto in fase di digitazione.

Vi potrà capitare, a volte di trovare nei listati di programmi per TI 99/4a alcuni caratteri sottolineati. La sottolineatura rappresenta una particolare codifica dei tasti di controllo: dovrete quindi premere il tasto indicato dal carattere insieme al tasto CONTROL.

Ad esempio, se trovate A dovrete battere CONTROL+A; se trovate —, dovrete battere CONTROL+SPAZIO, e così via.



# Calendario perpetuo

Che giorno della settimana siete nati? Quanti giorni siete vissuti? Quale sarà il calendario del mese di ottobre del prossimo anno? Usate il "calendario perpetuo" per scoprirlo.

Con questo programma potrete sapere il giorno della settimana di una qualsiasi data compresa tra il 1 gennaio del 1800 ed il 31 dicembre del 2800. In questo intervallo di date potrete inoltre sapere il numero di giorni che intercorre tra due giorni qualsiasi e stampare il

calendario di un qualsiasi mese.

Se non siete interessati direttamente a queste informazioni, potrete sempre estrarre dal programma le routine che compiono la validazione di una data o il calcolo dei giorni tra due date per poterle usare nei programmi che scriverete voi stessi.

Quindi quando dovrete controllare una data non... reinventate la ruota, ma usate questa routine.

```

1 HOME
3 VTAB 12: HTAB 11: INVERSE : PRINT
  "CALENDARIO PERPETUO": NORMAL

4 FOR I = 1 TO 800: NEXT I
10 REM ** CALENDARIO PERPETUO
50 B$ = " ": FOR I = 1 TO 100: B$ =
  B$ + " ": NEXT I
60 AS$ = "*": FOR I = 1 TO 100: AS$ =
  AS$ + "*": NEXT I
100 GOSUB 430
105 HOME
107 INVERSE
110 VTAB 1: HTAB 18: PRINT "1L"
220 VTAB 1: HTAB 18: PRINT "1L"
240 VTAB 3: HTAB 11: PRINT "CALE
  NDARIO PERPETUO"
260 NORMAL
270 VTAB 7: HTAB 1: PRINT "1. NU
  MBERO DEI GIORNI TRA DUE DAT
  E."
280 VTAB 9: HTAB 1: PRINT "2. GI
  ORNO DELLA SETTIMANA DI UNA
  DATA."
290 VTAB 11: HTAB 1: PRINT "3. C
  ALENDARIO DI OGNI MESE."
340 VTAB 13: HTAB 1: PRINT "4. F
  INE."
350 VTAB 15: HTAB 7: PRINT "SCEL
  TA: "
390 VTAB 17: HTAB 6: PRINT LEFT$(
  B$,40): VTAB 15: HTAB 28: PRINT
  " ": VTAB 15: HTAB 28
400 A$ = " ": GET A$: IF A$ = " " THEN
  GOTO 400
410 VTAB 15: HTAB 28: PRINT A$
415 IF A$ < "1" AND A$ < "2" AND A$ < "3" AND A$ <
  "4" THEN VTAB 17: HTAB 7
  : PRINT "1, 2, 3 O 4 PREGO":
  FOR I = 1 TO 1000: NEXT I: GOTO
  390

```

```

420 ON VAL (A$) GOSUB 940,1490,
  1910,2830
425 GOTO 350
430 REM ** INIZIALIZZAZIONE
440 DIM DS$(7),MS$(12),DS(12),NL
  (7)
445 DIM CL$(5,7)

450 DATA "DOMENICA"
460 DATA "LUNEDI'"
470 DATA "MARTEDI'"
480 DATA "MERCOLEDI'"
490 DATA "GIOVEDI'"
500 DATA "VENERDI'"
510 DATA "SABATO"
520 DATA "GENNAIO",31
540 DATA "FEBBRAIO",28
550 DATA "MARZO",31
580 DATA "APRILE",30
590 DATA "MAGGIO",31
600 DATA "GIUGNO",30
610 DATA "LUGLIO",31
620 DATA "AGOSTO",31
630 DATA "SETTEMBRE",30

640 DATA "OTTOBRE",31
660 DATA "NOVEMBRE",30
690 DATA "DICEMBRE",31
750 DATA 1900,2100,2200,2300,25
  00,2600,2700
830 FOR I = 1 TO 7: READ DS$(I):
  NEXT I
860 FOR I = 1 TO 12: READ MS$(I)
  ,DS(I): NEXT I
900 FOR I = 1 TO 7: READ NL(I): NEXT
  I
920 RETURN
940 REM ** GIORNI TRA DUE DATE
  **
950 VTAB 17: HTAB 3: AD$ = " "
970 INPUT "PRIMA DATA (GG/MM/AAA
  A): ";AD$

```



```

980 IF A0$ = "" THEN VTAB 17: HTAB
1: PRINT LEFT$ (B$,30): RETURN

990 GOSUB 3050: IF OK THEN 1090
1000 VTAB 19: HTAB 5: PRINT "DAT
A ERRATA. RIBATTERLA."
1020 FOR I = 1 TO 1000: NEXT I: VTAB
19: HTAB 5: PRINT LEFT$ (B$,
30)
1030 VTAB 17: HTAB 1: PRINT LEFT$
(B$,40): GOTO 950
1090 Y1 = Y:M1 = M:D1 = 0
1100 VTAB 19: HTAB 2:A0$ = ""
1120 INPUT "SECONDA DATA (GG/MM/
AAAA): ";AD$
1130 GOSUB 3050: IF OK THEN 1260

1170 VTAB 21: HTAB 5: PRINT "DAT
A ERRATA. RIBATTERLA."
1180 FOR I = 1 TO 1000: NEXT I: VTAB
21: HTAB 5: PRINT LEFT$ (B$,
40)
1190 VTAB 19: HTAB 1: PRINT LEFT$
(B$,40): GOTO 1100
1260 Y2 = Y:M2 = M:D2 = 0
1290 GOSUB 3190
1300 VTAB 21: HTAB 10: PRINT "IL
NUMERO DI GIORNI"
1305 VTAB 22: HTAB 7: PRINT "TRA
QUESTE DATE E' ";
1350 PRINT TOS
1370 VTAB 23: HTAB 5: PRINT "(PR
EMI UN TASTO PER CONTINUARE"
;
1380 A$ = "": GET A$: IF A$ = "" THEN
1380
1390 VTAB 17: HTAB 1: PRINT LEFT$
(B$,40): VTAB 19: HTAB 1: PRINT
LEFT$ (B$,40)
1392 VTAB 21: HTAB 1: PRINT LEFT$
(B$,40)
1395 VTAB 22: HTAB 1: PRINT LEFT$
(B$,40): VTAB 23: HTAB 1: PRINT
LEFT$ (B$,39): GOTO 950
1490 REM ** GIORNI DELLA SETTI
MANA
1500 Y1 = 1983:M1 = 1:D1 = 1
1530 VTAB 17: HTAB 3:A0$ = ""
1550 INPUT "DATA (GG/MM/AAAA): "
;AD$
1560 IF A0$ = "" THEN VTAB 17: HTAB
3: PRINT LEFT$ (B$,40): RETURN

1570 GOSUB 3050: IF OK THEN 1670
1580 VTAB 19: HTAB 3: PRINT "DAT
A ERRATA. RIBATTERLA."
1590 FOR I = 1 TO 1000: NEXT I
1620 VTAB 19: HTAB 3: PRINT LEFT$
(B$,40): VTAB 17: HTAB 3: PRINT
LEFT$ (B$,30): GOTO 1530
1670 Y2 = Y:M2 = M:D2 = 0
1700 GOSUB 3190
1710 IF TD > 32767 THEN TD = TD -
32767: GOTO 1710

1740 WD = INT ((TD / 7 - INT (T
D / 7)) * 7 + .05) * SGN (T
D / 7)

```

```

1750 IF PR = 0 THEN WD$ = DS$(7 -
WD): GOTO 1755
1751 IF WD > 0 THEN WD$ = DS$(WD
): GOTO 1755
1752 WD$ = DS$(7)
1755 VTAB 20: HTAB 5: PRINT "QUE
STO GIORNO E' UN ";
1760 INVERSE : PRINT WD$: NORMAL

1770 VTAB 22: HTAB 3: PRINT "(PR
EMI UN TASTO PER CONTINUARE)
";
1800 A$ = "": GET A$: IF A$ = "" GOTO
1800
1810 VTAB 17: HTAB 1: PRINT LEFT$
(B$,40): VTAB 19: HTAB 1: PRINT
LEFT$ (B$,40)
1815 VTAB 22: HTAB 1: PRINT LEFT$
(B$,40): VTAB 20: HTAB 1: PRINT
LEFT$ (B$,40)
1850 GOTO 1530
1910 REM ** ROUTINE DI STAMPA
CALENDARIO **
1930 Y1 = 1983:M1 = 1:D1 = 1
1931 VTAB 18: HTAB 5: PRINT "HAI
LA STAMPANTE (S/N)? ";: GET
T$
1932 VTAB 18: HTAB 34: PRINT T$
1933 IF T$ = "S" THEN 1938
1934 VTAB 20: HTAB 5: PRINT "LA
STAMPANTE E' NECESSARIA!!!"
1935 FOR I = 1 TO 1000: NEXT I
1936 FOR I = 17 TO 22: VTAB INT
(I): PRINT LEFT$ (B$,40): NEXT
I
1937 GOTO 350
1938 VTAB 19: INPUT "IN QUALE SL
OT E' INSERITA? ";N$
1939 IF N$ = "" OR N$ = "0" THEN
N$ = "1"
1940 N% = VAL (N$)
1980 VTAB 21: HTAB 5
1990 INPUT "DATA (MM/AAAA): ";A
0$
2000 IF A0$ = "" THEN VTAB 18: HTAB
1: PRINT B$: PRINT LEFT$ (B
$,40): RETURN
2020 A0$ = "0" + A0$:A0$ = RIGHT$
(A0$,7)
2030 AD$ = "01/" + AD$
2040 GOSUB 3050: IF OK THEN 2140

2050 VTAB 22: HTAB 1: PRINT "DAT
A ERRATA. RIBATTERLA."
2060 FOR I = 1 TO 1000: NEXT I
2070 HTAB 1: VTAB 21: PRINT LEFT$
(B$,40): HTAB 1: VTAB 22: PRINT
LEFT$ (B$,40): GOTO 1980
2140 Y2 = Y:M2 = M:D2 = 0
2150 GOSUB 3190
2180 IF TO > 32767 THEN TO = TO -
32767: GOTO 2180
2190 WD = INT ((TO / 7 - INT (T
D / 7)) * 7 + .05) * SGN (T
D / 7)
2200 IF PR = 0 THEN WD = 7 - WD:
GOTO 2250
2210 IF WD = 0 THEN WD = 7

```

```

2230 VTAB 23: HTAB 5: PRINT "PRE
MI UN TASTO PER STAMPARE ";
2250 A$ = "": GET A$: IF A$ = "" THEN
2250
2251 VTAB 23: PRINT LEFT$ (B$,4
0);
2260 PR# N%: PRINT CHR$ (9);"80
N";
2270 MX = DS(M2): IF LL AND M2 =
2 THEN MX = 29
2280 DM = 0
2290 FOR I = 1 TO 5
2300 FOR J = 1 TO 7
2310 IF ((I - 1) * 7) + J < WD OR
DM + 1 > MX THEN CL$(I,J) =
"*" + LEFT$ (B$,10): GOTO 2
325
2320 DM = DM + 1:DM$ = STR$ (DM)
:CL$(I,J) = "*" + DM$ + LEFT$
(B$, (9 - LEN (DM$)))
2325 NEXT J
2330 NEXT I
2340 ST = 0
2345 IF DM = MX THEN 2380
2350 FOR I = DM + 1 TO MX
2360 ST = ST + 1:CL$(5,ST) = LEFT$
(CL$(5,ST),4) + "/" + STR$
(I) + LEFT$ (B$,4)
2370 NEXT I
2380 LN = LEN (MS$(M2)) + LEN (
STR$ (Y2)) + 1
2390 LN = INT ((80 - LN) / 2): PRINT
CHR$ (12): PRINT LEFT$ (B$
, LN);MS$(M2);" "; STR$ (Y2)
2405 PRINT
2440 PRINT " "; LEFT$ (A$,78)
2490 GOSUB 3480: PRINT " ";
2500 FOR I = 1 TO 7:LD = LEN (D
S$(I))
2505 S1 = 10 - LD:S2 = INT (S1 /
.2):S3 = S1 - S2
2510 PRINT "*";: IF S2 < > 0 THEN
PRINT LEFT$ (B$,S2);
2520 PRINT DS$(I);: IF S3 < > 0
THEN PRINT LEFT$ (B$,S3);
.
2530 NEXT I
2540 PRINT "*": GOSUB 3480
2560 GOSUB 3540
2570 FOR I = 1 TO 5
2580 PRINT " ";
2590 FOR J = 1 TO 7: PRINT CL$(I
,J);
2600 NEXT J: PRINT "*"
2630 FOR J = 1 TO 5: GOSUB 3480:
NEXT J
2660 IF I < 5 THEN GOSUB 3540: NEXT
I
2670 PRINT " "; LEFT$ (A$,78)
2715 PRINT CHR$ (12): PR# 0
2720 VTAB 23: HTAB 5: PRINT "(PR
EMI UN TASTO PER CONTINUARE)
";
2730 A$ = "": GET A$: IF A$ = "" THEN
2730
2740 VTAB 21: HTAB 1: PRINT LEFT$
(B$,40): VTAB 23: HTAB 1: PRINT
LEFT$ (B$,39)
2750 VTAB 20: HTAB 1: PRINT LEFT$
(B$,40)
2800 GOTO 1980
2830 HOME
2835 END
3050 REM ** VALIDAZIONE DATA
3060 OK = 0
3070 AL = LEN (AD$): IF AL = 10 THEN
3090
3075 IF MID$ (AD$,2,1) = "/" THEN
AD$ = "0" + AD$:AL = AL + 1:
IF AL = 10 THEN 3090
3080 IF MID$ (AD$,5,1) = "/" THEN
AD$ = LEFT$ (AD$,3) + "0" +
RIGHT$ (AD$,6)
3090 IF MID$ (AD$,3,1) < > "/"
THEN RETURN
3092 IF MID$ (AD$,6,1) < > "/"
THEN RETURN
3120 Y = VAL (RIGHT$ (AD$,4)): IF
Y < 1800 OR Y > 2800 THEN RETURN
3130 LP = 0: IF Y / 100 ( ) INT
(Y / 100) THEN IF Y / 4 = INT
(Y / 4) THEN LP = - 1
3140 IF Y = 2000 OR Y = 2400 OR
Y = 2800 THEN LP = - 1
3145 M = VAL (MID$ (AD$,4,2)): IF
M < 1 OR M > 12 THEN RETURN
3150 MX = DS(M): IF M = 2 AND LP THEN
MX = 29
3160 D = VAL (LEFT$ (AD$,2)): IF
D < 1 OR D > MX THEN RETURN
3170 OK = - 1: RETURN
3190 REM ** CALCOLA I GIORNI T
RA DUE DATE **
3210 PR = 0:TD = 0
3220 IF Y1 < Y2 THEN PR = - 1
3240 IF Y1 = Y2 AND M1 < M2 THEN
PR = - 1
3250 IF Y1 = Y2 AND M1 = M2 AND
D1 < D2 THEN PR = - 1
3260 L1 = 0: IF Y1 / 100 < > INT
(Y1 / 100) THEN IF Y1 / 4 =
INT (Y1 / 4) THEN L1 = - 1
3265 IF Y1 = 2000 OR Y1 = 2400 OR
Y1 = 2800 THEN L1 = - 1
3270 L2 = 0: IF Y2 / 100 ( ) INT
(Y2 / 100) THEN IF Y2 / 4 =
INT (Y2 / 4) THEN L2 = - 1
3275 IF Y2 = 2000 OR Y2 = 2400 OR
Y2 = 2800 THEN L2 = - 1
3280 FY = Y1:FM = M1:FD = D1:FL =
L1:LY = Y2:LM = M2:LD = D2:L
L = L2
3282 IF PR = 0 THEN FY = Y2:FM =
M2:FD = D2:FL = L2:LY = Y1:L
M = M1:LD = D1:LL = L1
3290 NY = LY - FY: IF NY > 0 THEN
TD = INT (NY * 365.25) - 36
5
3310 FOR I = 1 TO 7
3320 IF NL(I) > FY AND NL(I) < L
Y THEN TD = TD - 1

```

```

3330 NEXT I
3335 IF LM - 1 = 0 THEN 3345
3340 FOR I = 1 TO LM - 1:TD = TD
    + DS(1): NEXT I
3345 IF NY < > 2 AND NY < > 3 THEN
    3370
3347 FOR I = FY + 1 TO LY - 1: IF
    I / 100 < > INT (I / 100) THEN
    IF I / 4 = INT (I / 4) THEN
    TD = TD + 1
3350 IF I = 2000 OR I = 2400 OR
    I = 2800 THEN TD = TD + 1
3355 NEXT I
3370 IF LL AND LM > 2 THEN TD =
    TD + 1
3380 TD = TD + LD
3389 IF FM = 12 THEN 3420
3390 FOR I = FM + 1 TO 12:TD = T
    D + DS(1): NEXT I


```

```

3420 TD = TD + DS(FM) - FD
3430 IF FL AND FM < 3 THEN TD =
    TD + 1
3440 IF FY < > LY THEN RETURN
3445 IF LL THEN TD = TD - 366: RETURN
3450 TD = TD - 365: RETURN
3470 REM ** ROUTINES DI STAMPA
    CALENDARIO **
3480 PRINT " ";
3490 FOR K = 1 TO 7: PRINT "*" +
    LEFT$(B$,10);: NEXT K
3520 PRINT "*"
3530 RETURN
3540 PRINT " "; LEFT$(A$,78): RETURN
3550 FOR K = 1 TO 7: PRINT LEFT$
    (A$,11);: NEXT K
3560 RETURN

```

## Tic Tac Toe



Extended basic

Suppongo che ad ognuno di voi sia capitato, almeno una volta, di misurarsi contro qualcuno in quel classico gioco di ragionamento chiamato tris.

... Bene. Ora avete grazie a questo programma la possibilità di misurare la vostra abilità con quella che il vostro Texas avrà raggiunto una volta che avrete caricato completamente questo listato nella sua memoria.

Il gioco è stato realizzato con 3 diversi livelli di difficoltà, ognuno dei quali è più completo co-

me numero di strategie attacco-difesa, di quello che lo precede. Inoltre il giocatore può scegliere se iniziare per primo o per secondo (è evidente che chi inizia ha in alcuni casi, notevoli vantaggi)

Naturalmente il calcolatore effettua anche un controllo sulla correttezza delle mosse del giocatore e rileva un'eventuale vittoria od un pareggio.

Le istruzioni per far apprendere al computer la vostra mossa sono date da programma.

```

0 REM *****
100 REM * TIC-TAC-TOE *
120 REM * per *
    * TI-99/4A *
130 REM *****
140 REM ti-basic
    copyright Massimo Bianchi
150 CALL CLEAR
160 CALL COLOR(15,2,2)
170 CALL COLOR(14,16,16)
180 CALL SCREEN(6)
190 FOR H=2 TO 13
200 CALL COLOR(H,16,4)
210 NEXT H
220 CALL COLOR(1,4,4)

```

```

230 GOSUB 2070
240 REM ** PRESENTAZIONE ED IS
    TRUZIONI **
250 DISPLAY TAB(3);"MASSIMO BIAN
    CHI SOFTWARE":::::TAB(10);"prese
    nta":::::
260 DISPLAY TAB(13);"THE":::::TAB
    (4);"T I C - T A C - T O E":::::
    ;
270 FOR H=30 TO 2 STEP -.5
280 CALL SOUND(-60,-5,H)
290 NEXT H
300 CALL SOUND(-2000,-6,0)
310 FOR H=1 TO 30
320 CALL SCREEN(16)

```



```

330 CALL SCREEN(6)
340 NEXT H
350 GOSUB 2070
360 PRINT TAB(2);"VUOI LE ISTRUZ
IONI? (S/N)":::::::::::
370 CALL KEY(3,K,K)
380 IF (K<>ASC("S"))*(K<>ASC("N"
))THEN 370
390 CALL SOUND(-200,330,2)
400 GOSUB 2070
410 IF K=ASC("N")THEN 510
420 PRINT " QUESTA E'UNA VERSION
E "; " COMPUTERIZZATA DEL FAMOSO"
;" GIOCO:":TAB(4);"T I C - T A
C - T O E":
430 PRINT " IL PROGRAMMA E'STATO
IDEATO": " PER PERMETTERE AD UN
GIO-": " CATORE DI GAREGGIARE CON
": " IL COMPUTER."
440 PRINT " LO SCOPO E'ALLINEARE
": " TRE PEDINE IN SENSO:": " ORI
ZZONTALE,VERTICALE": " E OBLIQUO"

450 PRINT " PER INSERIRE LA MOSS
A,": " BASTA BATTERE IL NUMERO ":
" CHE CORRISPONDE ALLA CASEL-": "
LA SCELTA."
460 PRINT ::TAB(2);"PREMI UN TA
STO PER GIOCARE"
470 CALL KEY(3,K,K)

480 IF K=-1 THEN 470
490 GOSUB 2070
500 REM GOTO 2620
510 REM ** INIZIO DEL SOTTOPRO
GRAMMA PRINCIPALE **
520 CALL COLOR(16,13,4)
530 CALL CHAR(159,"FFFFFFFFFFFF
FFF")
540 PRINT TAB(5);"INIZI TU? (S/
N)":::::::::::
550 CALL KEY(3,J,J)
560 IF (J<>ASC("S"))*(J<>ASC("N"
))THEN 550
570 CALL SOUND(-100,350,2)
580 GOSUB 2070
590 PRINT "INSERISCI IL LIVELLO
GIOCO:"::::::"1) FACILE"::::::"2)
MEDIO"::::::"3) DIFFICILE":::::
:::
600 CALL KEY(3,YY,YY)
610 YY=YY-48
620 IF (YY<>1)*(YY<>2)*(YY<>3)TH
EN 600
630 CALL SOUND(-100,300,2)
640 GOSUB 2070
650 PRINT TAB(5);"T I C - T A C
- T O E"::::::::::::TAB(
8);"MOSSA:":
660 MOS=0
670 FOR H=6 TO 18 STEP 4
680 CALL HCHAR(H,10,159,13)
690 NEXT H

```

```

700 FOR H=10 TO 22 STEP 4
710 CALL VCHAR(6,H,159,13)
720 NEXT H
730 RANDOMIZE
740 START=1
750 FOR A=8 TO 16 STEP 4
760 FOR B=12 TO 20 STEP 4
770 CALL HCHAR(A,B,48+START)
780 START=START+1
790 NEXT B
800 NEXT A
810 REM ** RICEVIMENTO DELLE M
OSSE DEL GIOCATORE **
820 IF J=ASC("N")THEN 1010
830 CALL CHAR(154,"007E7E7E7E7E
E")
840 CALL HCHAR(23,18,63)
850 CALL KEY(3,K,S)
860 IF MOS=9 THEN 2750
870 IF (K<49)+(K>57)THEN 940
880 IF F(K-48)<>0 THEN 840
890 CALL SOUND(-200,110,0)
900 CALL HCHAR(23,18,K)
910 VALORE=1
920 GOSUB 2160
930 GOTO 970
940 CALL HCHAR(23,18,32)
950 DELAY@VAR=ABS(-1)
960 GOTO 840

970 F(K-48)=1
980 V=1
990 GOSUB 1240
1000 V=-1
1010 IF YY=1 THEN 1130
1020 GOSUB 1620
1030 IF YY=2 THEN 1130
1040 GOSUB 2920
1050 V=-1
1060 GOSUB 1460
1070 IF (MOS=1)*(F(5)=0)THEN 113
0
1080 IF (MOS=0)+(MOS>1)THEN 1130

1090 FF=(INT(RND*5)*2+1)
1100 IF FF=5 THEN 1090
1110 F(FF)=-1
1120 GOTO 2050
1130 IF F(5)<>0 THEN 1160
1140 F(5)=-1
1150 GOTO 2810
1160 GOSUB 1460
1170 V=1
1180 GOSUB 1240
1190 GOSUB 1460
1200 V=-1
1210 GOSUB 1240
1220 GOSUB 1460
1230 GOSUB 1960
1240 FOR H=0 TO 6 STEP 3
1250 IF (F(H+1)=V)*(F(H+2)=V)*(F
(H+3)=V)THEN 1320
1260 NEXT H

```

```

1270 FOR H=0 TO 2
1280 IF (F(H+1)=V)*(F(H+4)=V)*(F(H+7)=V) THEN 1320
1290 NEXT H
1300 IF ((F(1)=V)*(F(5)=V)*(F(9)=V))+((F(3)=V)*(F(5)=V)*(F(7)=V))) THEN 1320
1310 RETURN
1320 IF V=1 THEN 1400
1330 GOSUB 2840
1340 L$=" HO VINTO IO!!"
1350 B=09
1360 C=23
1370 GOSUB 2120
1380 GOTO 2580
1390 GOTO 2810
1400 L$=" HAI VINTO TU!"
1410 GOSUB 2160
1420 C=23
1430 B=09
1440 GOSUB 2120
1450 GOTO 2580
1460 REM ** CONTROLLO SE AL GIOCATORE MANCA UNA CASELLA **
1470 FOR H=0 TO 6 STEP 3
1480 IF (F(H+1)=V)*(F(H+2)=V)*(F(H+3)=0) THEN 1760
1490 IF (F(H+2)=V)*(F(H+3)=V)*(F(H+1)=0) THEN 1780
1500 IF (F(H+2)=0)*(F(H+1)=V)*(F(H+3)=V) THEN 1800
1510 NEXT H
1520 FOR H=0 TO 2
1530 IF (F(H+1)=V)*(F(H+4)=V)*(F(H+7)=0) THEN 1820
1540 IF (F(H+1)=0)*(F(H+4)=V)*(F(H+7)=V) THEN 1840
1550 IF (F(H+4)=0)*(F(H+1)=V)*(F(H+7)=V) THEN 1860
1560 NEXT H
1570 IF (F(9)=0)*(F(1)=V)*(F(5)=V) THEN 1880
1580 IF (F(1)=0)*(F(5)=V)*(F(9)=V) THEN 1900
1590 IF (F(7)=0)*(F(3)=V)*(F(5)=V) THEN 1920
1600 IF (F(3)=0)*(F(5)=V)*(F(7)=V) THEN 1940
1610 RETURN
1620 REM ** CASO DELLA PEDINA IN UN LATERALE **
1630 IF (F(1)=0)*(F(2)=1)*(F(3)=0)*(F(4)=1)*(F(7)=0) THEN 1680
1640 IF (F(1)=0)*(F(2)=1)*(F(3)=0)*(F(6)=1)*(F(7)=0) THEN 1700
1650 IF (F(3)=0)*(F(6)=1)*(F(9)=0)*(F(8)=1)*(F(7)=0) THEN 1720
1660 IF (F(1)=0)*(F(4)=1)*(F(7)=0)*(F(8)=1)*(F(9)=0) THEN 1740
1670 RETURN
1680 F(1)=-1
1690 GOTO 2050

```

```

1700 F(3)=-1
1710 GOTO 2050
1720 F(9)=-1
1730 GOTO 2050
1740 F(7)=-1
1750 GOTO 2050
1760 F(H+3)=-1
1770 GOTO 2050
1780 F(H+1)=-1
1790 GOTO 2050
1800 F(H+2)=-1
1810 GOTO 2050
1820 F(H+7)=-1
1830 GOTO 2050
1840 F(H+1)=-1
1850 GOTO 2050
1860 F(H+4)=-1
1870 GOTO 2050
1880 F(9)=-1
1890 GOTO 2050
1900 F(1)=-1
1910 GOTO 2050
1920 F(7)=-1
1930 GOTO 2050
1940 F(3)=-1
1950 GOTO 2050
1960 REM ** IL COMPUTER POSIZIONA LA PEDINA SENZA ADIACENZE
1970 RE=INT(RND*9)+1
1980 IF MOS>=9 THEN 2750
1990 IF F(RE) THEN 1970
2000 F(RE)=-1
2010 V=-1
2020 GOSUB 1240
2030 GOTO 2050
2040 GOTO 840
2050 GOSUB 1240
2060 GOTO 2810
2070 REM ** PULIZIA VERTICALE DELLO SCHERMO **
2080 CALL VCHAR(1,1,31,48)
2090 CALL VCHAR(1,31,31,48)
2100 CALL VCHAR(1,3,32,672)
2110 RETURN
2120 FOR H=1 TO LEN(L$)
2130 CALL HCHAR(C,B+H-1,ASC(SEG$(L$,H,1)))
2140 NEXT H
2150 RETURN
2160 REM ** STAMPA DELLA MOSSA **
2170 ON K-48 GO SUB 2190,2220,2250,2280,2310,2340,2370,2400,2430
2180 IF VALORE=1 THEN 2460 ELSE 2520
2190 R=7
2200 C=11
2210 RETURN
2220 R=7
2230 C=15
2240 RETURN

```

```

2250 R=7
2260 C=19
2270 RETURN
2280 R=11
2290 C=11
2300 RETURN
2310 R=11
2320 C=15
2330 RETURN
2340 R=11
2350 C=19
2360 RETURN
2370 R=15
2380 C=11
2390 RETURN
2400 R=15
2410 C=15
2420 RETURN
2430 R=15
2440 C=19
2450 RETURN
2460 REM ** STAMPA DELLA PEDINA
    DEL GICATORE **
2470 CALL HCHAR(R,C+1,149)
2480 MOS=MOS+1
2490 CALL HCHAR(R+1,C,149,3)
2500 CALL HCHAR(R+2,C+1,149)
2510 RETURN
2520 CALL HCHAR(R,C,141,3)
2530 CALL HCHAR(R+1,C,141)
2540 CALL HCHAR(R+1,C+2,141)
2550 CALL HCHAR(R+2,C,141,3)
2560 CALL HCHAR(R+1,C+1,32)
2570 RETURN
2580 REM *** FINE DEL GIOCO ***

2590 FOR H=1 TO 800
2600 NEXT H
2610 CALL HCHAR(23,3,32,28)
2620 L$=" VUOI GIOCARE ANCORA?(S
/N)"
2630 B=3
2640 GOSUB 2120

```

```

2650 CALL KEY(3,K,S)
2660 IF (K<>ASC("S"))*(K<>ASC("N
"))THEN 2650
2670 IF K=ASC("S")THEN 2700
2680 CALL CLEAR
2690 STOP
2700 FOR H=1 TO 9
2710 F(H)=0
2720 NEXT H
2730 GOSUB 2070
2740 GOTO 510
2750 CALL HCHAR(23,3,32,28)
2760 L$="IL GIOCO E`PARI"
2770 C=23
2780 B=9
2790 GOSUB 2120
2800 GOTO 2580
2810 REM ** STAMPA DELLA MOSSA
    DEL COMPUTER **
2820 GOSUB 2840
2830 GOTO 840
2840 CALL SOUND(-100,220,0)
2850 MOS=MOS+1
2860 FOR H=1 TO 9
2870 IF (F(H)=0)+(F(H)=1)THEN 29
    00
2880 ON H GO SUB 2190,2220,2250,
    2280,2310,2340,2370,2400,2430
2890 GOSUB 2520
2900 NEXT H
2910 RETURN
2920 IF (F(1)=1)*(F(2)=0)*(F(3)=
    0)*(F(4)=0)*(F(5)=-1)*(F(6)=0)*(
    F(7)=0)*(F(8)=0)*(F(9)=1)THEN 29
    50
2930 IF (F(1)=0)*(F(2)=0)*(F(3)=
    1)*(F(4)=0)*(F(5)=-1)*(F(6)=0)*(
    F(7)=1)*(F(8)=0)*(F(9)=0)THEN 29
    50
2940 RETURN
2950 F(6)=-1
2960 GOTO 2050
2970 END

```



Extended basic

## Attack!

Questo è un gioco da provare prima di addentrarsi in tutte le più o meno sofisticate battaglie spaziali di cui Papersoft vi ha fornito numerosi listati.

Il listato è estremamente semplice ed è costruito sull'idea-base del comandante stellare che tor-

na dopo uno stressante combattimento, nel corso del quale ha esaurito le munizioni, ed improvvisamente viene attaccato da una miriade di astronavi-kamikaze aliene che gli si dirigono contro. C'è poco da decidere sulla strategia da seguire a questo punto. l'unica cosa



sensata è darsela a gambe!

Ed è qui che entri in gioco tu: nei panni dello scaloganto comandante devi dirigere l'astronave a forma di scodella alla larga dagli alieni (utilizzando i tasti -S- e -D-) cercando di resistere

il più a lungo possibile.

Inizialmente il computer vi darà la possibilità di scegliere tra 5 differenti livelli di gioco, dopodiché inizierà la vostra gara contro il tempo. Buona fortuna!

```
100 ! *****
110 ! **      ATTACK!      **
120 ! *****
130 ! TI-99      EXT. BASIC
140 ! PROGRAMMA BASE
150 DIM R(16)
160 CALL CLEAR
170 INPUT "LIVELLO DI GIOCO(1-5)
? 2:LE
180 CALL MAGNIFY(2)
190 SC=0
200 CALL COLOR(3,16,1)
210 CALL COLOR(4,16,1)
220 A$="003C7EDB7E3C0000"
230 B$="0000000081423C00"
240 CALL CHAR(42,A$)
250 CALL CHAR(95,B$)
260 CALL CLEAR
270 CALL SCREEN(2)
280 CALL SPRITE(#20,95,16,150,75
)
290 FOR K=3 TO 16
300 R(K)=INT(RND*10+2*LE)
310 CALL SPRITE(#K,42,K,16,17*K-
41,R(K),0)
320 GOSUB 540
330 NEXT K
340 FOR K=3 TO 16
```

```
350 GOSUB 540
360 IF M=-1 THEN K=16
370 SC=SC+1
380 R(K)=R(K)+LE
390 CALL MOTION(#K,R(K),0)
400 DISPLAY AT(3,3):SC
410 CALL SOUND(5,440,5)
420 NEXT K
430 IF M<>-1 THEN GOTO 340
440 CALL SOUND(2000,-7,5)
450 CALL SCREEN(12)
460 CALL DELSPRITE(ALL)
470 CALL COLOR(3,2,1)
480 CALL COLOR(4,2,1)
490 DISPLAY AT(20,2):"RITENTI (S/
N)?"
500 ACCEPT AT(20,15):Q$
510 IF SEG$(Q$,1,1)="S" OR SEG$(
Q$,1,1)="s" THEN GOTO 160
520 END
530 ! MOVIMENTO NAVI
540 CALL KEY(0,X,Y):: IF X=83 TH
EN CALL MOTION(#20,0,-12)ELSE IF
Y=0 THEN 540
550 IF X=68 THEN CALL MOTION(#20
,0,12)
560 CALL COINC(ALL,M)
570 RETURN
```

## Capogiro

Questo gioco è certamente ispirato alla roulette: due giocatori devono infatti scegliere un numero compreso tra 0 e 41 e scommettere somme a piacere entro il vincolo di bilancio rappresentato dal proprio capitale (inizialmente 100 sterline); chi azzecca il numero vince considerevoli somme di denaro, mentre ovviamente che sbaglia previsione vede assottigliarsi il suo

capitale. La partita termina quando almeno uno dei due giocatori resta privo di fondi. Le differenze con la roulette sono tuttavia notevoli (grafica a parte): la più notevole consiste nel fatto che, scelto un numero, al giocatore viene assegnato un range di 5 numeri (quello scelto e 4 adiacenti) con i quali può vincere. Dopo aver battuto il programma e averlo sal-



16/48K

vato con GO TO 9000, lanciatelo: vi verranno richiesti i nomi dei due giocatori, la somma che intendete puntare, i numeri prescelti. A questo punto, dopo essere stati informati sui rispettivi range di vittoria, partirà un disco che toccherà ad una ad una tutte le caselle numerate, arrestandosi solo dopo un discreto numero di giri. Chi avrà previsto correttamente (e fortunosamente!) il numero estratto si vedrà accreditare una discreta somma sul suo conto. Va notato che non è possibile far combaciare i numeri (o i range) prescelti dai due

giocatori: in altri termini i due concorrenti non possono vincere simultaneamente. Il programma può servire da interessante pretesto per studiare il problema delle sicure sugli input: come è noto spesso i programmi di chi è alle prime armi sono carenti sotto questo profilo, e si arrestano in relazione a input scorretti battuti dall'utente. Provate dunque ad esaminare il funzionamento di queste sicure, particolarmente numerose in questo listato e rivolte a salvaguardare lo svolgimento del gioco sotto numerosi profili.

```

1 PAPER 7: BORDER 7: INK 0:
  BRIGHT 0: FLASH 0: OVER 0:
  CLS
3 RESTORE 30
5 LET CO=0: LET ODD1=1: LET O
  DD2=1: LET RANGE1=1: LET RA
  NGE2=1: LET RANGE3=1: LET R
  ANGE4=1: LET VITT=0
10 POKE 23658,8
20 FOR F=USR "A" TO USR "C"+7:
  READ A: POKE F,A: NEXT F
30 DATA 255,129,129,129,129,12
  9,129,255
31 DATA 255,128,128,128,128,12
  8,128,255
32 DATA 255,1,1,1,1,1,1,255
40 FOR N=1 TO 41: READ A,B,C
50 PRINT OVER 1;AT A,B;CHR$ C
  ;AT A,B;N
60 IF N>=10 THEN PRINT AT A,B
  ;CHR$ 145;CHR$ 146;AT A,B;
  OVER 1;N
65 NEXT N
70 DATA 1,1,144,1,3,144,1,5,14
  4,1,7,144
71 DATA 1,9,144,1,11,144,1,13,
  144,1,15,144
72 DATA 1,17,144,1,19,145,1,22
  ,145,1,25,145
73 DATA 1,28,145,3,28,145,5,28
  ,145,7,28,145
74 DATA 9,28,145,11,28,145,13,
  28,145,15,28,145
75 DATA 17,28,145,19,28,145,21
  ,28,145,21,25,145
76 DATA 21,22,145,21,19,145,21
  ,16,145,21,13,145
77 DATA 21,10,145,21,7,145,21,
  4,145,21,1,145
78 DATA 19,1,145,17,1,145,15,1
  ,145,13,1,145

```

```

79 DATA 11,1,145,9,1,145,7,1,1
  45,5,1,145,3,1,145
80 GO TO 200
100 RESTORE 70: FOR N=1 TO 41:
  READ A,B,C
105 LET PA=RND*2+.35
106 IF PA<.5 AND CO>1 AND N>=10
  THEN PRINT AT A,B;
  PAPER 4; INK 1; FLASH 1;
  CHR$ 145;CHR$ 146;AT A,B;N:
  LET CO=0: GO TO 400
107 IF PA<.5 AND CO>1 AND N<10
  THEN PRINT AT A,B;
  PAPER 4; INK 1; FLASH 1;
  CHR$ C;AT A,B;N: LET CO=0:
  GO TO 400
108 IF PA<.5 AND CO<=1 THEN
  LET PA=.5
110 IF N>=10 THEN PRINT AT A,B
  ; PAPER 2; FLASH 1;CHR$ 145
  ;CHR$ 146;AT A,B;N: PAUSE P
  A: PRINT AT A,B; FLASH 0;
  PAPER 7;CHR$ 145;CHR$ 146;
  AT A,B; OVER 1;N
130 PRINT OVER 1; PAPER 2;
  FLASH 1;AT A,B;CHR$ C;AT A
  ,B;N: PAUSE PA: PRINT AT A,
  B; FLASH 0; PAPER 7;CHR$ C;
  OVER 1;AT A,B;N
140 NEXT N
150 LET CO=CO+1
160 GO TO 100
200 INPUT "NOME GIOCATORE 1: ";
  P$
203 IF P$="" THEN GO TO 200
205 IF LEN P$>=10 THEN LET P$=
  P$( TO 10)
210 INPUT "NOME GIOCATORE 2: ";
  L$

```

```

213 IF L$="" THEN GO TO 210
215 IF LEN L$>=10 THEN LET L$=
    L$( TO 10)
220 LET M1=100: LET M2=100
230 GO SUB 600
231 PRINT AT 3,4;P$;" HA E ";M1
    ;AT 5,4;L$;" HA E ";M2
240 INPUT "PUNTATA GIOC.1= ";ST
    AKE1
245 IF STAKE1>M1 THEN GO TO 24
    0
246 IF STAKE1<=0 THEN GO TO 24
    0
250 INPUT "PUNTATA GIOC.2= ";ST
    AKE2
255 IF STAKE2>M2 THEN GO TO 25
    0
256 IF STAKE2<=0 THEN GO TO 25
    0
257 LET M1=M1-STAKE1: LET M2=M2
    -STAKE2
260 INPUT "GIOC.1: NUMERO= ";GU
    1
265 IF GU1>41 OR GU1<1 THEN
    GO TO 260
270 IF GU1>2 AND GU1<40 THEN
    LET RANGE1=GU1+2: LET RANG
    E2=GU1-2
280 IF GU1<=2 THEN LET RANGE1=
    GU1+4: LET RANGE2=GU1
290 IF GU1>=40 THEN LET RANGE2
    =GU1-4: LET RANGE1=GU1
300 INPUT "GIOC.2: NUMERO= ";GU
    2
305 IF GU2>41 OR GU2<1 THEN
    GO TO 300
310 IF GU2>2 AND GU2<40 THEN
    LET RANGE3=GU2+2: LET RANG
    E4=GU2-2
320 IF GU2<=2 THEN LET RANGE3=
    GU2+4: LET RANGE4=GU2
330 IF GU2>=40 THEN LET RANGE4
    =GU2-4: LET RANGE3=GU2
340 IF (GU2>=RANGE2 AND GU2<=RA
    NGE1) OR (RANGE4<=RANGE1
    AND RANGE4>=RANGE2) OR (RA
    NGE3<=RANGE1 AND RANGE3>=RA
    NGE2) THEN GO TO 300
350 PRINT AT 7,4;P$;"=";" PUNTA
    TA ";STAKE1;AT 9,4;L$;"=";"
    PUNTATA ";STAKE2
360 PRINT AT 11,4;"-";P$;AT 12,
    4;"VINCE CON NUMERI";AT 1
    3,4;"COMPRESI FRA ";RANGE1;
    " E ";RANGE2;AT 15,4;"-";L$

```

```

;AT 16,4;"VINCE CON NUMER
I";AT 17,4;"COMPRESI FRA ";
RANGE3;" E ";RANGE4
380 GO TO 100
400 BEEP .3,10: IF N<=RANGE1
    AND N>=RANGE2 THEN LET M1
    =M1+STAKE1+(STAKE1*ODD1):
    LET CHI=0: LET D$=P$:
    GO SUB 800
410 IF N<=RANGE3 AND N>=RANGE4
    THEN LET M2=M2+STAKE2+(ST
    AKE2*ODD2): LET CHI=4:
    LET D$=L$: GO SUB 800
420 PAUSE 100: FOR N=3 TO 19:
    PRINT AT N,4;"
    ": NEXT N:
    GO TO 425+VITT
425 PRINT AT 10,6;"NESSUNO HA
    VINTO";AT 12,9;"QUESTA MAN
    CHE"
430 PAUSE 100: FOR N=3 TO 19:
    PRINT AT N,4;"
    ": NEXT N:
    LET VITT=0
435 IF M1<=0 OR M2<=0 THEN
    GO TO 450
440 GO TO 230
450 IF M1<=0 AND M2<=0 THEN
    CLS : PRINT AT 9,9;
    FLASH 1;"CHE SFORTUNA !!!"
    ; FLASH 0;AT 11,6;"AVETE PE
    RSO ENTRAMBI !!!": GO TO 460
451 IF M2<=0 THEN CLS :
    PRINT AT 9,0;"CHE SFORTUNA
    ";L$'"HAI ESAURITO I SOLDI
    E HAI PERSO";AT 12,0;"CONG
    RATULAZIONI ";P$'"HAI VINTO
    !!! (E ";M1;" )"
452 IF M1<=0 THEN CLS :
    PRINT AT 9,0;"CHE SFORTUNA
    ";P$'"HAI ESAURITO I SOLDI
    E HAI PERSO";AT 12,0;"CONG
    RATULAZIONI ";L$'"HAI VINTO
    !!! (E ";M2;" )"
460 INPUT "UN'ALTRA PARTITA (S/
    N)? ";W$
470 IF W$="S" THEN RUN
480 IF NOT W$="S" THEN STOP
600 RESTORE 610: FOR N=4 TO 14
    STEP 2: READ DATI: PRINT
    AT N,4;CHR$ 145;" ";
    CHR$ 146;AT N,5;"1:";DATI:
    NEXT N
610 DATA 1,2,4,6,8,10

```



```

615 RESTORE 610: FOR N=1 TO
  INT (RND*7): READ ODD1:
  NEXT N: RESTORE 610: FOR M
  =1 TO INT (RND*7): READ ODD
  2: NEXT M
620 PRINT AT 18,4;"GIOC.1=";"1:
  ";ODD1;AT 19,4;"GIOC.2=";"1
  ";ODD2
630 PAUSE 150: FOR N=4 TO 19:
  PRINT AT N,4;"
  ": NEXT N:
  RETURN
800 PRINT AT 11+CHI,5; FLASH 1;
  DS;AT 19,4; FLASH 0;"COMPLI
  MENTI,HAI VINTO !": LET VIT
  T=10: FOR N=1 TO 255: OUT 1

```

```

32,N: BEEP .00175,RND*65:
  NEXT N: PRINT AT 18,6;"
  "
810 RETURN
8999 STOP
9000 CLS : SAVE "CAPOGIRO"
  LINE 1
9100 CLS : PRINT AT 10,0;"RIAVVO
  LGI IL NASTRO & CONNETTI <
  EAR > PER VERIFICARE."
9200 VERIFY "CAPOGIRO"
9300 CLS : PRINT AT 13,14;
  FLASH 1;"O.K."
9400 STOP

```

## Grandi stampe



16/48K

Per sfruttare questo programma è necessario disporre di una stampante. Chi non la possiede può togliersi la curiosità di vedere cosa fa 'Grandi stampe' sostituendo alla linea 470 lo statement LPRINT con PRINT. Inutile dire che il risultato sarà meno duraturo che non realizzando una scritta su carta (come è infatti noto "Display volant, scripta manent"...). Il programma presenta alcune interessanti peculiarità: innanzitutto contiene un convertitore di numeri decimali in binario che consente di 'girare' di 90 gradi verso il basso un qualsiasi UDG. In secondo luogo esso realizza la stampa degli UDG girati nel senso di scorrimento della ZX Printer ed ingranditi 24 volte. Questa caratteristica può dunque consentire all'utente di stampare, notevolmente ingranditi, i caratteri grafici da lui creati (ad esempio per corredare un articolo). Inoltre, poiché di default l'area RAM che conserva gli UDG contiene le

prime 21 lettere maiuscole dell'alfabeto inglese, potrete anche stampare giganteschi messaggi. Unico inconveniente: mancano in tal modo le lettere V, W, X, Y, Z. Per porre rimedio a tale situazione potete modificare il programma in modo che punti il set di caratteri standard (risiede in ROM a partire dalla locazione 15616); oppure potete definire, più semplicemente, un UDG non usato in stampa in modo che assuma l'aspetto della lettera mancante. Ancora meglio: non stampate messaggi contenenti quelle lettere!

Un'ultima precisazione: una volta lanciato il programma vi viene richiesto di scegliere tra l'opzione 'Stampa' e l'opzione 'Codici Binari'; quest'ultima non ha alcuna utilità pratica, ma è stata inserita per mostrarvi come il programma 'capovolge' i codici degli UDG per consentirne la stampa nel senso della carta.

```

1 REM stampa ingrandita
10 PRINT AT 2,0;"VUOI :""")
  STAMPARE MESSAGGIO SU CARTA

```

```

2) VEDERE CODICI BINARI G
  IRATI"
20 IF INKEY$="1" THEN LET bin
  =0: GO TO 100

```

```

30 IF INKEY$="2" THEN LET bin
   =450: GO TO 100
40 GO TO 20
100 REM
101 REM CONVERSIONE IN BINARIO
102 REM
110 BORDER 7: PAPER 7: INK 0:
   CLS
150 DIM a(8): LET a(1)=128:
   LET a(2)=64: LET a(3)=32:
   LET a(4)=16: LET a(5)=8:
   LET a(6)=4: LET a(7)=2:
   LET a(8)=1
160 INPUT "MESSAGGIO > ";a$
170 FOR s=1 TO LEN a$: IF a$(s)
   =" " THEN GO TO 190
180 IF CODE a$(s)<65 OR CODE a$(s)>85
   OR CODE a$(s)>117 AND
   CODE a$(s)<144 OR CODE a$(s)
   >164 THEN PRINT a$(s);" N
   ON E' UN UDG !": BEEP 1,-20
   : STOP
190 NEXT S
200 FOR p=1 TO LEN a$
210 IF a$(p)=" " THEN CLS :
   COPY : GO TO 490
220 LET y=USR a$(p)
230 DIM b$(8,8)
240 FOR i=0 TO 7
250 LET b$(i+1)="00000000"
260 LET x=PEEK (y+i)
270 FOR n=1 TO 8
280 IF x>=a(n) THEN LET b$(i+1
   ,n)="1": LET x=x-a(n)

```

```

290 NEXT n
300 NEXT i
310 REM
311 REM GIRA IL MESSAGGIO
312 REM
340 DIM e$(8,8)
350 LET k=1
360 FOR i=1 TO 8
370 LET f$=""
380 FOR n=8 TO 1 STEP -1
390 LET f$=f$+b$(n,k): NEXT n
400 LET e$(i)=f$
410 LET k=k+1: NEXT i
420 REM
421 REM STAMPA GRANDI LETTERE
422 REM
440 GO TO 450+bin
450 FOR i=1 TO 8
460 LET c$="": FOR n=1 TO 8
470 LET c$=c$+(" " AND e$(i,
   n)="0")+("{4SG8}" AND e$(i,
   n)="1")
480 NEXT n: LPRINT c$'c$'c$:
   NEXT i
490 NEXT p
500 BEEP 1,20: STOP
900 REM
901 REM CODICI BINARI GIRATI
902 REM
910 FOR n=1 TO 8: PRINT "BIN ";
   e$(n): NEXT n
920 PRINT : NEXT p
950 BEEP 1,20: STOP

```

## Caccia al tesoro

**C64**

Joystick

Scopo di questo gioco è recuperare un tesoro lasciato dai pirati su un'isola sperduta, evitando una serie di ostacoli. Il tesoro è composto da monete e lingotti d'oro e da un forziere pieno di gioielli. Gli ostacoli sono barili di rum, sabbie mobili e fantasmi di antichi pirati (che appaiono sotto forma di teschi); insomma, tutto

ciò che può mettere in pericolo la vita di un cacciatore di tesori. Quando, guidando il cacciatore col joystick n. 2, sarete riusciti a raccogliere la maggior parte delle monete e dei lingotti d'oro (potete lasciare sul terreno una moneta e/o un lingotto), accederete a un nuovo schermo.

Non è necessario recuperare il forziere, ma se lo fate guadagnate molti più punti; ricordate però di tener premuto il pulsante del joystick mentre cercate di attraversare le sabbie mobili che circondano il forziere. Durante il gioco potete vedere il numero della schermata e il

punteggio.

Ci sono cinque livelli di difficoltà: quello che selezionate determina il numero di ostacoli fissi e mobili, oltre ad influenzare il punteggio guadagnato per ogni tesoro che recuperate.

```

100 POKE53280,6:POKE53281,1
110 POKE56,48:POKE52,48:CLR
120 PRINT"{CLR}{ 7 GIU' }{<1>}***
*****{RVS} CACCIA AL TESO
RO {OFF}*****"
130 PRINT"{ 7 GIU' }{BLU}
{ 8 SPAZI}RIDEFINIZIONE
{ 2 SPAZI}CARATTERI"
140 POKE56334,PEEK(56334)AND254:
POKE1,PEEK(1)AND251
150 FORI=0TO511:POKEI+12288,PEEK
(53248+I):NEXT
160 POKE1,PEEK(1)OR4:POKE56334,P
EEK(56334)OR1
170 FORI=12288+35*8TO12288+47*8+
7:READA:POKEI,A:NEXT
180 FORI=12288+58*8TO12288+61*8+
7:READA:POKEI,A:NEXT
190 POKE53272,21:GOSUB1100:SN=54
272:POKESN+24,15:POKESN+5,17
:POKESN+6,240
200 POKESN,100:GOSUB420
210 POKE53272,(PEEK(53272)AND240
)OR12:RN=RN+1:P=1:Q=40
220 GOSUB710:GOSUB1330:GOSUB1450
230 CL=1902
240 JS=PEEK(56320)AND15
250 TL=CL:Z=CL:ONJS-4GOSUB360,34
0,350,410,380,400,390,410,37
0,330,410
260 CL=Z:POKESN+1,50:POKESN+4,33
:GOSUB460
270 IFFLANDNOTDFTHENGOSUB990
280 POKESN+4,32:IFDFTHEN1760
290 POKETL,32:POKETL+Q,32
300 POKECL,46:POKECL+Q,47:POKECM
+CL,3:POKECM+Q+CL,3:GOSUB780
:IFGC<2ANDGB<2THEN210
310 GOSUB860:IFDFTHEN1760
320 GOTO240
330 Z=Z-Q:RETURN
340 Z=Z-39:RETURN
350 Z=Z+P:RETURN
360 Z=Z+41:RETURN
370 Z=Z+Q:RETURN
380 Z=Z+39:RETURN
390 Z=Z-P:RETURN
400 Z=Z-41:RETURN
410 RETURN

```

```

420 PRINT"{GIU' }{RVS}{<5>}LIVELL
O DI DIFFICOLTA' (1-5)?"
430 GETA$:IFA$=""THEN430
440 AA=VAL(A$):IFAA<1ORAA>5THEN4
30
450 RETURN
460 :
470 TC=0:IFPEEK(CL)=35ORPEEK(CL+
Q)=35ORPEEK(CL)=36ORPEEK(CL+
Q)=36THENTC=P
480 IFPEEK(CL)=37ORPEEK(CL+Q)=37
ORPEEK(CL)=38ORPEEK(CL+Q)=38
THENTC=P
490 IFPEEK(CL)=39ORPEEK(CL+Q)=39
ORPEEK(CL)=40ORPEEK(CL+Q)=40
THENTC=P
500 IFPEEK(CL)=41ORPEEK(CL+Q)=41
THENTC=P
510 IFTCTHENCL=TL:RETURN
520 IF(PEEK(CL)=47ANDPEEK(CL+Q)=
32)OR(PEEK(CL)=32ANDPEEK(CL+
Q)=46)THENRETURN
530 IFPEEK(CL)=32ANDPEEK(CL+Q)=3
2THENRETURN
540 IFPEEK(CL)=46ANDPEEK(CL+Q)=4
7THENRETURN
550 IFPEEK(CL)=42ORPEEK(CL+Q)=42
THEN610
560 IFPEEK(CL)=61ORPEEK(CL+Q)=61
THENGGB=GB-P:SC=SC+10*AA:POKE
SN+1,30:POKESN+4,33
570 IFPEEK(CL)=60ORPEEK(CL+Q)=60
THENGCG=GC-P:SC=SC+AA:POKESN+
1,80:POKESN+4,33
580 IFPEEK(CL)=43ORPEEK(CL+Q)=43
ORPEEK(CL)=44ORPEEK(CL+Q)=44
THEN640
590 IFPEEK(CL)=45ORPEEK(CL+Q)=45
THEN990
600 RETURN
610 :
620 POKETL,32:POKETL+Q,32:POKECL
,46:POKECL+Q,47:POKECL+CM,0:
POKECL+CM+Q,0
630 GOTO1060
640 :
650 D=INT(RND(1)*10)+1
660 JB=NOT(-(PEEK(56320)AND16)/1
6)

```



```

670 IFJBANDPEEK(CJ)=59ANDD>2THEN
    CL=CH:POKECI,32:SC=SC+100*AA
    :FG=P:RETURN
680 IFJBANDPEEK(CJ)<>59ANDD>2THE
    NCL=CL+47:GOTO460
690 CL=CH+41:POKETL,32:POKETL+Q,
    32:POKECL,46:POKECM+CL,3:GOT
    O1060
700 :
710 PRINT"{CLR}":CM=54272
720 FORI=1024TO1063:POKEI,35:POK
    ECM+I,14:NEXT
730 FORI=1103TO2023STEP40:POKEI,
    35:POKECM+I,14:NEXT
740 FORI=1982TO1943STEP-1:POKEI,
    35:POKECM+I,14:NEXT
750 FORI=1984TO1064STEP-40:POKEI
    ,35:POKECM+I,14:NEXT
760 RETURN
770 :
780 PRINT"{HOME}{" 24 GIU'}{DES}
    [<1>]SCHERMO"RN"PUNTI"SC;:IF
    FGTHEN810
790 RD=INT(RND(1)*15)
800 ONRDGOSUB810,810,810,810,810
    ,810,820,810,810,810,810,810
    ,840,810,810
810 RETURN
820 IFPEEK(CI)=32ANDPEEK(CJ)=32T
    HENPOKECI,58:POKECJ,59:POKEC
    I+CM,6:POKECJ+CM,6
830 RETURN
840 IFPEEK(CI)=58THENPOKECI,32:P
    OKECJ,32
850 RETURN
860 :
870 D=INT(RND(1)*AA)+1:ONDGOSUB9
    40,950,960,970,980
880 TS=SK:Z=SK:POKESK,32
890 ONINT(RND(1)*8)+1GOSUB330,34
    0,350,360,370,380,390,400
900 SK=Z:IFPEEK(SK)=32THENONDGOS
    UB1610,1620,1630,1640,1650:G
    OTO930
910 IFPEEK(SK)=46ORPEEK(SK)=47TH
    EN610
920 SK=TS
930 POKESK,42:POKESK+CM,7:RETURN

940 SK=S1:RETURN
950 SK=S2:RETURN
960 SK=S3:RETURN
970 SK=S4:RETURN
980 SK=S5:RETURN
990 :
1000 POKETL,32:POKETL+Q,32:POKEC
    L,46:POKECL+Q,47:POKECL+CM,
    4:POKECL+CM+Q,4
1010 POKESN+1,40:POKESN+4,33:FOR
    I=1TO10:NEXT:POKESN+1,45:FO
    RI=1TO25:NEXT
1020 POKESN+4,32

```

```

1030 TL=CL:Z=CL:ONINT(RND(1)*8)+
    1GOSUB330,340,350,360,370,3
    80,390,400
1040 CL=Z:IFFL=.THENFL=P:GOTO460

1050 IFFL=PTHENFL=.:GOTO460
1060 :
1070 POKESN+1,40:POKESN+4,33:FOR
    X=1TO5:POKESN+1,40:FORL=1TO
    50:NEXT:POKESN+1,20
1080 FORL=1TO50:NEXT:NEXT:POKESN
    +4,32
1090 DF=1:RETURN
1100 PRINT"{CLR}{RVS}{<4>}SEI AL
    LA CACCIA DI UN ANTICO TESO
    RO LA- ";
1110 PRINT"SCIATO DAI PIRATI SU
    UN'ISOLA SPERDUTA. ";
1120 PRINT"{BLU}GUIDA IL CACCIAT
    ORE CON IL JOYSTICK IN
    { 2 SPAZI}";
1130 PRINT"PORTA 2 FINO ALLE MON
    ETE, AI LINGOTTI{ 3 SPAZI}
    ";
1140 PRINT"D'ORO E AL FORZIERE D
    EL TESORO.{ 9 SPAZI}";
1150 PRINT"QUEST'ULTIMO E' CIRCO
    NDATO DA SABBIE MO-";
1160 PRINT"BILI. HAI IL 70% DI P
    ROBABILITA' DI PAS-";
1170 PRINT"SARE SE PREMI IL PULS
    ANTE MENTRE LE AT- ";
1180 PRINT"TRAVERSI.{ 31 SPAZI}
    ";
1190 PRINT"{PUR}IL TESORO E' MAL
    EDETTO E GLI SPIRITI DE-";
1200 PRINT"GLI ANTICHI PIRATI LO
    FANNO SCOMPARIRE
    { 2 SPAZI}";
1210 PRINT"OGNI TANTO.
    { 29 SPAZI}";
1220 PRINT"{GRN}GLI SPIRITI MALI
    GNI HANNO ANCHE SISTEMA-";
1230 PRINT"TO BARILI DI RUM SULL
    'ISOLA PER DISTRAR-";
1240 PRINT"RE I CACCIATORI.
    { 24 SPAZI}";
1250 PRINT"{<7>}SE IL CACCIATORE
    BEVE IL RUM, BARCOLLA
    { 2 SPAZI}";
1260 PRINT"E TU NON PUOI GUIDARL
    O.{ 17 SPAZI}"
1270 PRINT"{RED}{RVS}TABELLA DEI
    PUNTEGGI:{ 19 SPAZI}";
1280 PRINT"MONETE{ 3 SPAZI}= 1 X
    LIVELLO DI DIFFICOLTA'
    { 3 SPAZI}";
1290 PRINT"LINGOTTI = 10 X LIVEL
    LO DI DIFFICOLTA'
    { 2 SPAZI}";
1300 PRINT"FORZIERE = 100 X LIVE
    LLO DI DIFFICOLTA' ";

```

```

1310 RETURN
1320 :
1330 TB=INT(RND(1)*8)+2:PRINT"
      {HOME}":I=INT(RND(1)*3)+1:F
      ORX=0TOI:PRINT"{GIU'}":NEXT

1340 PRINTTAB(TB)"{GRN}&% '&% '"
1350 PRINTTAB(TB)"($&% ')&% '"
1360 PRINTTAB(TB)"{DES}$($&% '$)"

1370 PRINTTAB(TB)"{ 3 DES}$($)$"
1380 PRINTTAB(TB)"{ 3 DES}$ $ $"
1390 PRINTTAB(TB)"{ 5 DES}$[BLK]
      { 2 SPAZI},+"
1400 PRINTTAB(TB)"{GRN}{ 5 DES}
      ${BLU}:; {BLK},"
1410 PRINTTAB(TB)"{ 8 DES},+"
1420 PRINTTAB(TB)"{ 5 DES}+,+,+"
1430 CH=PEEK(209)+256*PEEK(210)-
      (160-(TB+7)):CJ=CH+Q:CI=CJ-
      P
1440 RETURN
1450 :
1460 FORI=PTO7*AA
1470 X=INT(RND(1)*720)+1064
1480 IFPEEK(X)<>32ORPEEK(X+P)<>3
      2ORPEEK(X-P)<>32ORPEEK(X+Q)
      <>32THEN1470

1490 IFPEEK(X-Q)<>32THEN1470
1500 POKE X,45:POKECM+X,4:NEXT
1510 FORI=PTO6*AA
1520 X=INT(RND(1)*720)+1064
1530 IFPEEK(X)<>32ORPEEK(X+P)<>3
      2ORPEEK(X-P)<>32ORPEEK(X+Q)
      <>32THEN1520
1540 IFPEEK(X-Q)<>32THEN1520
1550 POKE X,42:POKECM+X,0:NEXT
1560 FORI=PTOAA
1570 SK=INT(RND(1)*720)+1064:IFP
      EEK(SK)<>42THEN1570
1580 POKESK+CM,7
1590 ONIGOSUB1610,1620,1630,1640
      ,1650
1600 NEXT:GOTO1660
1610 S1=SK:RETURN
1620 S2=SK:RETURN
1630 S3=SK:RETURN
1640 S4=SK:RETURN
1650 S5=SK:RETURN
1660 GC=25:B=60:FORI=1TO25
1670 GOSUB1750
1680 IFPEEK(X)<>32ORPEEK(X+Q)><3
      2ORPEEK(X-Q)><32THEN1670
1690 POKE X,B:POKECM+X,7:NEXT
1700 GB=5:B=61:FORI=1TO5
1710 GOSUB1750
1720 IFPEEK(X)<>32ORPEEK(X+Q)><3
      2ORPEEK(X-Q)><32THEN1710
1730 POKE X,B:POKECM+X,7:NEXT
1740 RETURN

```

```

1750 X=INT(RND(1)*755)+1064:RETU
      RN
1760 :
1770 POKESN+4,32:POKE53272,21
1780 PRINT"{CLR}{ 3 GIU'}{RVS}
      {YEL}{ 9 SPAZI}UN'ALTRA VIT
      TIMA DELLA{ 9 SPAZI}";
1790 PRINT"{ 7 SPAZI}MALEDIZIONE
      DEI PIRATI !!!{ 7 SPAZI}"
1800 PRINT"{GIU'}{RVS}{RED}LIVEL
      LO DI DIFFICOLTA':"AA
1810 PRINT"{GIU'}{RVS}{RED}SCHER
      MI:"RN
1820 PRINT:PRINT"{RVS}{RED}PUNTI
      : "SC
1830 PRINT:PRINT"{RVS}{RED}GIOCH
      I ANCORA (S/N)?"
1840 GETAS:IFAS=""THEN1840
1850 IFAS="S"THENRN=0:FL=0:FG=0:
      DF=0:SC=0:PRINT"{CLR}":GOTO
      200
1860 END
1870 DATA255,255,255,255,255,255
      ,255,255,28,28,28,28,28,28,
      28,28
1880 DATA129,227,247,255,255,255
      ,255,156,15,127,127,227,143
      ,63,113,243
1890 DATA248,254,254,227,249,252
      ,207,227,199,159,62,56,120,
      96,96,64,121,120,28
1900 DATA14,6,6,7,
1910 DATA60,126,90,126,126,36,60
      ,24,28,63,31,255,255,127,12
      4,56
1920 DATA60,254,255,255,243,120,
      56,60,28,62,62,62,62,62,62,
      28
1930 DATA60,126,219,255,102,60,2
      4,255,189,189,189,189,36,36
      ,36,102
1940 DATA15,31,63,63,63,63,63,63
      ,248,252,254,254,254,254,25
      4,254
1950 DATA,,,24,24,,,,,,127,127,
      ,,0

```





# Pulcino

Ecco un gioco per bambini: al povero pulcino crolla il mondo addosso... È proprio il caso di dirlo poiché gli cadono addosso dei pezzi di cielo, spinti casualmente dal vento. L'unica difesa consiste nel lanciare contro di essi qualche sasso mandandoli in frantumi. Mano a mano

che si procede nel gioco, i pezzi cadranno con sempre maggiore velocità rendendo il confronto più emozionante. Il programma è diviso in due parti che devono essere registrate in sequenza su cassetta in modo che la prima parte possa caricare automaticamente la seconda.

## programma I

```
0 IFPEEK(7344)=PEEK(7344+25600) THEN4
1 POKE52,28:POKE56,28:CLR:FORT=7168TO7679:POKET,PEEK(T+25600):NEXT
4 PRINTCHR$(147)
5 GOSUB10000:GOSUB6000
30 PRINT:PRINT"ATTENDI CHE IL GIOCO{ 2 SPAZI}SIA CARICATO."
40 POKE198,5:POKE631,78:POKE632,69:POKE633,87:POKE634,13:POKE635,131:END
6000 PRINTCHR$(147);SPC(7);CHR$(18);"PULCINO";CHR$(146)
7000 PRINT:PRINT"TU GIOCHI LA PARTE DELPULCINO."
7001 PRINT
7010 PRINT"TI MUOVI NELLA ZONA{ 3 SPAZI}BASSA DELLO SCHERMO."
7011 PRINT
7020 PRINT"PEZZI DI CIELO (SPINTI CASUALMENTE DAL VENTO) TI CADONO ADDOSSO."
7030 PRINT:PRINT"PER FAR PUNTI, LANCIA UN SASSO CONTRO IL{ 4 SPAZI}PEZZO CHE CADE PER EMEN-"
7031 PRINT"{SU}DO LA BARRA DI SPAZIO."
7040 PRINT:PRINT"{SU}PUOI ANCHE COLPIRE I{ 2 SPAZI}PEZZI FERMI."
7050 PRINT:PRINTCHR$(18);"PREMI UN TASTO";CHR$(146)
7055 GETA$:IFA$=""THEN7055
7056 PRINTCHR$(147)
7060 PRINT"USA I TASTI I,M,J & K PER ANDARE SU, GIU',{ 2 SPAZI}A SINISTRA, A DESTRA.":PRINT
```

```
7080 PRINT"IL TUO PUNTEGGIO E IL NUMERO DEI PULCINI RI-MASTI APPAIONO NELLA"
7085 PRINT"PARTE ALTA DELLO{ 6 SPAZI}SCHERMO."
7090 PRINT:PRINT"OTTERRAI 50 PUNTI COL-PENDO I PEZZI DI CIELO CHE CADONO,"
7100 PRINT"E 10 PUNTI PER QUELLI FERMI."
7690 PRINT:PRINTCHR$(18);"PREMI UN TASTO";CHR$(146)
7700 GETA$:IFA$=""THEN7700
7701 PRINTCHR$(147)
7702 PRINT"OGNI VOLTA CHE COMPLETI LO SCHERMO E PASSI AL LIVELLO SUPERIORE,"
7703 PRINT"I PEZZI DI CIELO CADONO PIU' VELOCI."
7704 PRINT:PRINTCHR$(18);
7705 PRINT"SE PIU' DI 8 PEZZI{ 4 SPAZI}GIUNGONO A TERRA, IL{ 2 SPAZI}GIOCO FINISCE."
7711 PRINT:PRINT:PRINT:PRINT:PRINT
7720 FORT=1TO25:PRINTCHR$(145);SPC(4);CHR$(18);"BUONA FORTUNA!":FORP=1TO200:NEXTP
7730 PRINTCHR$(145);SPC(4);"BUONA FORTUNA!":FORP=1TO200:NEXTP:NEXTT
7740 PRINTCHR$(147)
9180 RETURN
10000 READA:IFA<0THENRETURN
10010 FORI=ATO A+7:READJ:POKEI,J:NEXT
10020 GOTO10000
10031 DATA7168,48,44,100,131,129,195,36,24
```



```

10041 DATA7176,12,18,233,33,241,
      14,9,16
10050 DATA7184,0,0,28,34,36,24,0
      ,0
10060 DATA7200,36,227,1,0,0,195,
      65,82
10070 DATA7216,0,0,0,0,0,1,3,133
10080 DATA7224,16,8,4,2,1,0,0,1

```

## programma 2

```

0 QW=4:POKE7678,4:POKE7675,9
1 TY=3:POKE7677,0:DIMC(30)
2 POKE7679,3:TY$=":3"
3 POKE36869,255:PS=PEEK(7675):ZZ
  =30720
4 PRINTCHR$(147)
5 TY=PEEK(7679):TY$=CHR$(58)+CHR
  $(TY+48)
6 QW=PEEK(7678)
7 V=36878:S1=36874:S2=36875:S3=3
  6876:S4=36877
11 POKE36879,8:X=11:Y=20
12 PRINTCHR$(147)
15 DEFFNA(Z)=INT(RND(1)*Z)
16 FORT=1TO30:C(T)=FNA(352)+7701
  :POKEC(T),0:POKEC(T)+ZZ,6:NEX
  T
17 X=11:Y=20
18 IFQW=1THENM=1:GOTO20
19 FORM=1TOQW
20 A=((Y*22)+X)+7680
21 PRINTCHR$(159);CHR$(19);"PUNT
  I:"SC"
22 PRINTCHR$(19);CHR$(156);SPC(1
  3);"PULCINI:";CHR$(157);TY$
30 POKEA,1:POKEA+1,6:POKEA+22,7:
  POKEA+23,10:POKEA+ZZ,7:POKEA+
  ZZ+1,7:POKEA+ZZ+22,7
31 POKEA+ZZ+23,7
40 IFPEEK(197)<>64THENGOSUB9000
50 IFF=1THENGOSUB9040
55 IFM<>1THEN65
56 IFR1=1THENGOSUB2170
57 IFR=31THEN7000
60 IFR1=0THENGOSUB2160
65 IFQW=1THEN18
2000 NEXTM:GOTO18
2160 R=R+1:IFR=31THEN7000
2162 R1=1:IFC(R)=0ORPEEK(C(R))=3
  2THEN2160
2165 Q=C(R):QQ=6
2166 IFQ1<>0THENQ1=32:QQ=0
2170 POKEQ,Q1:POKEQ+ZZ,QQ:Q=Q+22
  +FNA(3)-1:IFQ=AORQ=A+1ORQ=A
  +22ORQ=A+23THEN3100
2171 Q1=PEEK(Q)
2180 POKEQ,0:POKEQ+ZZ,6

```

```

10090 DATA7248,121,1,1,2,252,68,
      130,1
10100 DATA7352,1,1,6,4,10,5,2,1
10110 DATA7272,192,48,8,12,9,58,
      228,8
10120 DATA7360,41,42,60,72,164,1
      35,68,56
10130 DATA7304,252,2,1,1,1,253,3
      ,1
19999 DATA-1

```

```

2190 IFQ>8185THENR1=0:POKEQ,32:P
  OKEQ+ZZ,0:C(R)=0:DE=DE+1
2195 IFQ=BTHENPOKEQ,4:POKEQ+ZZ,2

2196 IFDE=PSTHEN9200
2200 IFQ=BTHENFORT=170TO255:POKE
  V,15:POKES4,T:POKES2,T:POKE
  S1,T:POKES3,T:NEXT:POKEQ,32

2201 IFQ=BTHENPOKEQ+ZZ,0
2202 POKEV,0:POKES4,0:POKES1,0:P
  OKES3,0:POKES2,0
2210 IFQ=BTHENS1=SC+50:C(R)=0:R1
  =0:F=0
2222 RETURN
3100 POKEA,23:POKEA+1,13:POKEA+2
  2,24:POKEA+23,17:POKEA+ZZ,7
  :POKEA+ZZ+1,7:POKEA+ZZ+22,7

3101 POKEA+ZZ+23,7
3102 POKEV,15:POKES2,158:FORT=1T
  O600:NEXT:POKES2,0:FORT=1TO
  60:NEXT
3103 POKEV,15:POKES2,158:FORT=1T
  O300:NEXT:POKES2,0:FORT=1TO
  60:NEXT
3104 POKEV,15:POKES2,181:FORT=1T
  O1050:NEXT:POKES2,0:FORT=1T
  O60:NEXT
3105 POKEV,0:POKES1,0:POKES2,0
3200 TY=TY-1:IFTY=0THEN9200
3201 TY$=CHR$(58)+CHR$(TY+48)
3203 R1=0
3210 IFTY<>0THENF=0:POKEB,32:POK
  EA,32:POKEA+1,32:POKEA+22,3
  2:POKEA+23,32
3220 IFTY<>0THENPOKEB+ZZ,0:POKEA
  +ZZ,0:POKEA+1+ZZ,0:POKEA+22
  +ZZ,0:POKEA+23+ZZ,0:GOTO17
7000 PRINTCHR$(147):PRINT:PRINT:
  PRINT:PRINT:PRINT:PRINT:PRI
  NT
7001 POKE36869,240:FORT=1TO6:POK
  EV,15:POKES2,239
7002 PRINTCHR$(5);CHR$(145);SPC(
  3);CHR$(18);
7003 PRINT"LIVELLO SEGUENTE":FOR
  P=1TO150:NEXT:POKES2,0
7005 PRINTCHR$(145);SPC(3);"LIVE

```

```

      LLO SEGUENTE":FORP=1TO150:N
      EXTP:NEXTT
7006 POKE36869,255:PRINTCHR$(5)
7007 IFQW=1THENQW=2
7010 R=0:POKE7679,TY:QW=QW-1:POK
      E7678,QW
7011 IFPS=1THEN7020
7012 PS=PS-1
7020 R=0:DE=0:GOTO3
9000 J=PEEK(197):POKEA,32:POKEA+
      1,32:POKEA+22,32:POKEA+23,3
      2
9001 POKEA+ZZ,0:POKEA+ZZ+1,0:POK
      EA+ZZ+22,0:POKEA+23+ZZ,0
9003 IFF=1THEN9010
9005 IFJ=32THENF=1:B=A-22:FORT=1
      TO20:POKEV,15:POKES4,180:NE
      XT:POKES4,0
9010 IFJ=44THENX=X+1
9011 IFJ=20THENX=X-1
9015 IFX>21THENX=20
9016 IFX<0THENX=0
9020 IFJ=36THENY=Y+1

```

```

9021 IFJ=12THENY=Y-1
9025 IFY>21THENY=21
9026 IFY<17THENY=17
9030 RETURN
9040 POKEB,2:POKEB+ZZ,4
9050 POKEB,32:POKEB+ZZ,0:B=B-22
9070 IFB<7680THENF=0
9075 IFB=QTHEN2195
9080 IFPEEK(B)<>0THEN9090
9085 POKEB,4:FORT=225TO170STEP-1
      :POKEV,15:POKES4,T:POKES2,T
      :NEXT:SC=SC+10
9090 POKES2,0:POKES4,0:POKEV,0
9159 IFPEEK(B)<>32THENPOKEB,32:P
      OKEB+ZZ,0:F=0:GOTO9166
9165 POKEB,2:POKEB+ZZ,4
9166 :
9180 RETURN
9200 POKE36869,240:PRINTCHR$(147
      );CHR$(5);"IL CIELO E' CADU
      TO."
9205 PRINT"{GIU'}PUNTEGGIO:"SC
9210 POKE198,0

```

## Guitar



Quante volte hai segretamente invidiato i ragazzi che durante le vacanze riescono a conquistarsi amiche ed amici solamente perché sanno suonare una chitarra e conoscono le canzoni più in voga?

Il programma, che gira sul VIC 20 inespanso, è diviso in 2 parti. La prima ti permette di eseguire quella cosa che si deve sempre fare prendendo in mano la chitarra: accordarla. Non devi fare altro che riprodurre le tonalità che per ogni corda il tuo computer ti fa ascoltare. Una volta ben accordato lo strumento, puoi far girare la seconda parte del programma (che si carica automaticamente), la quale ti permette di vedere la posizione della mano sinistra e di sentire il suono prodotto dall'accordo da te scelto.

Puoi selezionare ben 36 accordi diversi, cioè quelli in tonalità maggiore, minore e settima su tutte le 12 note compresi i diesis, rispondendo alla domanda 'nuovo accordo' con il nu-

mero relativo alla nota seguito dal segno del tipo di accordo da te scelto, che puoi vedere in basso sullo schermo. A questo punto osserva la diteggiatura della mano sinistra sulla riproduzione della tastiera della chitarra: in alto vi è la corda che emette il suono più acuto (mi cantino) poi via via tutte le altre fino ad arrivare al mi basso (sesta corda); i numeri corrispondono alle dita della tua sinistra dall'indice al mignolo (1-4). Per eseguire l'accordo premi con il dito corrispondente il tasto indicato, cercando di tenere l'ultima falange perpendicolare al piano della tastiera; nel frattempo ascolta il suono prodotto dal tuo VIC e confrontalo con quello emesso dalla chitarra. Se vuoi riascoltare il suono premi 'SPACE', e per cambiare accordo premi un qualsiasi altro tasto seguito dal codice del nuovo accordo. Ti capiterà selezionando alcuni accordi di notare una fila di fronte a un "barré" che va eseguito distendendo l'indice sul tasto per premere contem-

poraneamente le 6 corde; essendo questi tra gli accordi più difficili, ti consigliamo di provare ad eseguirli quando avrai già un po' di pratica con gli accordi "normali".

Avrai già notato che accanto alla tastiera compaiono anche delle lettere: A, B, e, a volte, X; quest'ultima indica che la corda alla sua destra non va suonata, poiché non fa parte dell'accordo. A e B invece indicano l'"alternato" e il "basso", cioè le due corde che, pizzicate con il pollice della mano destra, emettono un

suon grave che si alterna in molti accompagnamenti a quello dell'accordo vero e proprio ottenuto pizzicando le prime 3 corde con medio, anulare e mignolo. Prova ed otterrai degli effetti tipici di molti generi musicali.

Il procedimento per salvare il programma è il seguente: digita e registra la prima parte con il nome di GUITAR, dopodiché dai una NEW, digita la seconda parte e salvala con un qualsiasi nome di file subito dopo la prima.

## programma I

```

5 POKE36879,25:POKE55,218:POKE56
,27:R=0
10 PRINT"{CLR}"TAB(4)"** ACCORDI
**"
11 PRINT"{RED}{ 5 GIU' }
{ 12 SPAZI}U*F*DI"
12 PRINT"{ 4 DES}##{ 6 SPAZI}G
{ 4 SPAZI}H"
13 PRINT"{BLU}{ 4 DES}{< 12 +>}
{RED} Y"
14 PRINT"{ 4 DES}##{ 6 SPAZI}G
{ 4 SPAZI}H"
15 PRINT"{ 12 SPAZI}J*D*FK{BLU}"

20 IFR=1THEN60
25 PRINT"{ 4 GIU' }VUOI ACCORDARE
LA CHI-TARRA? (S/N)";
26 GOSUB50
27 IFP=28THENPRINT" OK{ 3 SU}":G
OTO60
28 IFP=41THEN30
29 GOTO26
30 PRINT"{CLR}{GIU' }"TAB(5)"TAST
I:"
31 PRINT"{GIU' } 1 2 3 4 5 6"
32 FORK=1TO6:PRINT"[<Q>] *++*++*
++*++"K"CORDA":NEXT
33 POKE36878,10
34 B$="150175196210220229"
35 PRINT"{GIU' }AGENDO SULLA MECC
ANICA"
36 PRINT"CERCA DI OTTENERE UN
{ 2 SPAZI}SUONO SIMILE A QUEL
LO CHE STAI SENTENDO"
37 PRINT"PER PROSEGUIRE PREMI
{ 2 SPAZI}UN TASTO"
38 FORK=6TO1STEP-1:PRINT"{HOME}
{ 12 GIU' }DELLA "K" CORDA"
39 FORK=6TO1STEP-1:PRINT"{HOME}
{ 12 GIU' }DELLA "K" CORDA"
40 N=19-3*K:S$=MID$(B$,N,3):S=VA
L(S$)
41 POKE36875,S:FORI=1TO200:NEXT

```

```

42 GOSUB50
43 NEXT
44 POKE36875,0
45 PRINT"{CLR}VUOI RIPETERE L'AC
COR-DATURA? (S/N)":GOSUB50
46 IFP=41THEN29
47 IFP=28THENR=1:GOTO10
48 GOTO45
49 GOTO45
50 P=PEEK(197):IFP=64THEN50
55 RETURN

60 PRINT"{ 6 GIU' }*ATTENDI CON P
AZIENZA*"
80 FORK=7134TO7673:READA:POKEK,A
:NEXT
90 LOAD
100 DATA,,33,22,,11,,5,4,150,188
,202,210,222,229
101 DATA3,,53,54,42,,5,6,165,18
8,210,222,227,233
102 DATA,,33,22,34,11,,5,4,150,1
88,202,218,222,229
105 DATA4,,62,63,64,0,5,6,172,1
92,213,224,230,234
106 DATA4,,63,64,52,0,5,6,172,1
92,213,224,229,234
107 DATA4,,63,,64,,5,6,172,192,
213,220,230,234

110 DATA,9,,21,33,22,4,5,,175,1
96,215,225,232
111 DATA,9,,22,33,11,4,5,,175,1
96,215,225,230
112 DATA,9,,22,11,23,4,5,,175,1
96,215,222,232
115 DATA6,,82,83,84,,5,6,179,20
0,218,227,233,237
116 DATA6,,83,84,72,,5,6,179,20
0,218,227,232,237
117 DATA6,,83,,84,,5,6,179,200,
218,224,233,237
120 DATA,,22,23,11,,,6,5,150,183
,202,213,220,229
121 DATA,,22,23,,,6,5,150,183,2
02,210,220,229

```



```

122 DATA,,22,,11,34,,6,5,150,183
    ,202,213,225,229
130 DATA1,,33,34,22,,6,5,157,18
    8,205,215,222,230
131 DATA1,,33,34,,,6,5,157,188,
    205,213,222,230
132 DATA1,,33,,22,44,,6,5,157,18
    8,196,215,227,230
135 DATA2,,43,44,32,,6,5,160,19
    2,208,218,224,232
136 DATA2,,43,44,,,6,5,160,192,
    208,215,224,232
137 DATA2,,43,,32,54,,6,5,160,19
    2,202,218,229,232
140 DATA,32,21,,,34,6,5,165,183
    ,196,210,220,233
141 DATA3,,53,54,,,6,5,165,196,
    210,218,225,233
142 DATA,33,22,,,11,6,5,165,183
    ,196,210,220,230
145 DATA4,,63,64,52,,6,5,172,20
    0,213,222,227,234

```

```

146 DATA4,,63,64,,,6,5,172,200,
    213,220,227,234
147 DATA4,,63,,52,74,,6,5,172,20
    0,208,222,232,234
150 DATA,,,21,22,23,,5,6,150,175
    ,202,215,224,229
151 DATA,,,22,23,11,,5,6,150,175
    ,202,215,222,229
152 DATA,,,22,,23,,5,4,150,175,2
    02,210,224,229
155 DATA1,,,34,33,32,,5,6,157,17
    9,205,218,225,230
156 DATA1,,,33,34,22,,5,6,157,17
    9,205,218,224,230
157 DATA1,,,33,,34,,5,6,157,179,
    205,213,225,230
160 DATA2,,,42,43,44,,5,6,160,18
    3,208,220,227,232
161 DATA2,,,43,44,32,,5,6,160,18
    3,208,220,225,232
162 DATA,9,22,11,23,,24,5,4,,183
    ,208,215,227,232

```

## programma 2

```

100 Y=3:VX=7:LX=170:RX=800:PA=50
    0
140 A=36876:V=36878:DIMA%(5)
150 POKE36879,25:PRINT"{CLR}
    {GIU'}*ACCORDI PER CHITARRA*
    "
160 FORK=38510TO38641:POKEK,6:NE
    XT
170 FORK=7791TO7901STEP22:POKEK,
    235:NEXT
180 FORN=7793TO7903STEP22:FORK=N
    TON+16STEP2:POKEK,219:NEXT:N
    EXT
183 FORN=7792TO7902STEP22:FORK=N
    TON+18STEP2:POKEK,192:NEXT:N
    EXT
190 PRINT"{HOME}{ 12 GIU'}
    { 3 SPAZI}TASTI DA USARE:":P
    RINT:PRINT" [SPACE] PER RIPE
    TERE":PRINT
200 PRINT" [UN ALTRO TASTO] PER"
    :PRINT"{ 3 SPAZI}CAMBIARE AC
    CORDO":PRINT
210 PRINT"{RVS}1{OFF}DO"TAB(6)"
    {RVS}4{OFF}RE#"TAB(11)"{RVS}
    7{OFF}FA#"TAB(17)"{RVS}10
    {OFF}LA"
220 PRINT"{RVS}2{OFF}DO#"TAB(6)"
    {RVS}5{OFF}MI"TAB(11)"{RVS}8
    {OFF}SOL"TAB(17)"{RVS}11
    {OFF}LA":POKE8141,35:POKE388
    61,6

```

```

230 PRINT"{RVS}3{OFF}RE"TAB(6)"
    {RVS}6{OFF}FA"TAB(11)"{RVS}9
    {OFF}SOL#"TAB(17)"{RVS}12
    {OFF}SI"
240 PRINT"{RVS}+{OFF}MAGG"TAB(6)
    "{RVS}-{OFF}MIN"TAB(11)"
    {RVS}E{OFF}SETT"
250 GOSUB1030:FORK=1TO10:GETB$:N
    EXT:INPUT"{HOME}{ 2 GIU'}NUO
    VO ACCORDO";A$
260 RESTORE
270 S=VAL(A$):IFS<1ORS>12THEN250
280 FORK=1TOS:READB$:NEXT
290 R$=RIGHT$(A$,1):R=ASC(R$):IF
    R=43THENB$=B$+"+":N=0:GOTO32
    0
300 IFR=45THENB$=B$+"-":N=15:GOT
    O320
310 B$=B$+"7":N=30
320 GOSUB1030:PRINT"{HOME}
    { 2 GIU'}"TAB(8)B$
330 S=S*45+7089:S=S+N:GOSUB1000
340 P=PEEK(S):IFP=0THEN370
350 P=P*2+7768:FORK=PTOP+110STEP
    22
360 POKEK,49:NEXT
370 FORK=0TO5:GOSUB1050:IFP=0THE
    N410
380 IFP=9THENL=24:I=0:GOTO400
390 I=INT(P/10):L=P-I*10:L=L+48
400 N=7878-K*22+I*2:POKEN,L
410 NEXT
420 GOSUB1050:POKE7746+22*P,2

```



**A Milano,  
in via Mascheroni 14.  
La prima software  
libreria italiana**

Un tempo si andava in libreria per il gusto della scoperta, per il piacere di esser informati sulle novità. Per incontrarsi, discutere, chiedere un consiglio al libraio-amico. Tutto questo è ancora possibile, per un prodotto assolutamente nuovo: libri e riviste di informatica italiani ed esteri, software, giochi.

Dove? Alla **Libreria JACKSON**.  
La prima software - libreria italiana.

**A Milano, in via Mascheroni 14.**

**Tel. 02-437385**

**Vieni a trovarci: ti aspettiamo.**

```

430 GOSUB1050:POKE7746+22*P,1
440 FORK=0TO5
450 GOSUB1050:A%(K)=P:NEXT
455 FORJ=1TOY
460 POKEV,VX:FORK=0TO5:POKEA,A%(
    K)
470 FORL=1TOLX:NEXT:NEXT
480 FORL=1TORX:NEXT:POKEV,0
483 FORL=0TOPA:NEXT
485 NEXT
490 P=PEEK(197):IFP=32THEN455
500 IFP<>64THEN250
510 GOTO490
1000 FORK=7768TO7878STEP22:POKEK
    ,32:NEXT

```

```

1010 FORN=7770TO7880STEP22:FORK=
    NTON+18STEP2:POKEK,192:NEXT
    :NEXT
1020 RETURN
1030 FORK=7724TO7745:POKEK,32:NE
    XT
1040 :RETURN
1050 S=S+1:P=PEEK(S)
1060 RETURN
1100 DATADO,DO#,RE,RE#,MI,FA,FA#
    ,SOL,SOL#,LA,LA#,SI
1300 DATADO,DO#,RE,RE#,MI,FA,FA#
    ,SOL,SOL#,LA,LA#,SI

```

Per ricevere gli arretrati di **PAPER soft** compilare il seguente tagliando:

**J.soft** Compilare ed inviare in busta chiusa a:  
via Rosellini 12 - 20142 Milano - Tel. 02/6888228-683797-6880841-2-3

Nome \_\_\_\_\_

Cognome \_\_\_\_\_

Indirizzo \_\_\_\_\_

CAP \_\_\_\_\_ Città \_\_\_\_\_ Prov. \_\_\_\_\_

Desidero ricevere i seguenti numeri arretrati di

**PAPER soft** \_\_\_\_\_

a L. 2.000 cad. per un totale di L. .... a mezzo

☐ assegno allegato

☐ contanti allegati





# **Il Jacksoniano ha il Basic**

## **Video Basic, corso su cassetta per parlare subito**

Oggi è davvero facile imparare il Basic, con Video Basic il corso su cassetta che ti permette di programmare subito il tuo computer. È facile: tu chiedi, lui risponde, tu impari.

Passo dopo passo. Sul tuo schermo appaiono le domande, le risposte, gli esercizi

e tu, senza fatica, presto e bene, impari a dialogare col tuo computer, sia un VIC 20, un Commodore 64 o un Sinclair. Video Basic è in edicola. Provalo subito.

Oggi il Basic si impara così.



*facile*  
**in mano.**  
 col tuo computer.



**GRUPPO  
 EDITORIALE  
 JACKSON**

**IN EDICOLA  
 DALL' 8-1-85**

In omaggio  
 una fantastica cassetta giochi.

# VIDEO BASIC abbonarsi conviene

(5 splendidi raccoglitori  
insieme al corso completo)



Video Basic la trovi in  
edizionale a lire 8.000 il  
fascicolo con cassetta e  
manuale. Ma abbonarsi  
conviene; con 165.000  
lire avrai infatti il corso  
completo, a casa tua, e 5 splendidi  
(e pratici) raccoglitori del valore  
di 40.000 lire.  
**NON PERDERE L'OCCASIONE!**

## Desidero abbonarmi a Video Basic

- ☐ Per il computer Commodore VIC 20
- ☐ Per il computer Commodore 64
- ☐ Per il computer Sinclair Spectrum

Allega lire 165.000 con assegno n° \_\_\_\_\_ dello Banco \_\_\_\_\_ a allego fotocopia della ricevuta di  
versamento con voglio postale intestato a GRUPPO EDITORIALE JACKSON - MILANO, che mi dà diritto di ricevere a casa  
mia il corso completo e 5 raccoglitori.



Spedire a:  
**JACKSON**  
Via Rasellini, 12  
20124 Milano

Nome \_\_\_\_\_ Cognome \_\_\_\_\_

Via \_\_\_\_\_ N. \_\_\_\_\_

CAP \_\_\_\_\_ Città \_\_\_\_\_ Provincia \_\_\_\_\_